

MSX


POCKET BANK 12

ポケットバンク

BASICミニ講座

アクションゲーム38

ぐるーぶ・アレフ



ハラハラドキドキの連続……
手に汗にぎるアクション・ゲームが満載だっ！

アスキー出版局

MSX POCKET BANK 12

アクションゲーム38

ぐるーぷ・アレフ

アスキー出版局


C O N T E N T S

introduction

できるだけ短いプログラムで、アクションゲームの醍醐味を味わっていただく
と、38の面白アクションゲームをプログラム解説入りで収録したのが、このシ
ョートプログラム集です。
高いお金を払ってゲームソフトを買わなくても、そしてBASICの知識がゼン
ゼンなくても、MSXマシンさえあれば、手ごろなアクションゲームを学びなが
ら楽しめます。どのゲームから始めるかはキミの自由、サア、まずはMSXの前
に座ってキーをたたいてみよう。

4

●マンガ版
アクションゲームの
作り方




8

●スプライト機能
大図解



10


●グラフィック機能
大図解



12

●マンガ版
ゲームストーリー
発想法講座





16

プログラムテクニック ワンポイントレッスン



17

ショートプログラム

ACTION GAME

38

●硬くなった脳ミソをほくそう!
思考型アクションゲーム

- 1. 放火魔.....18
- 2. スカイダイビング.....20
- 3. バンク・パニック.....22
- 4. 酒樽ころりん.....24
- 5. 磁石の惑星.....26

●欲求不満のあなたに!
爽快シューティングゲーム

- 6. 手榴弾を投げろ!!.....28
- 7. タンク-2.....30
- 8. BOMB-X.....32
- 9. 星間大戦争.....34
- 10. 追撃、戦車戦!.....36
- 11. ベルリンの壁.....38
- 12. ハンティングゲーム.....40

●画面が動けば心もおどる!
スクロールゲーム

- 13. JUMP-JUMP.....42
- 14. 風船の襲来.....44
- 15. 鍾乳洞.....46
- 16. ドライブゲーム.....48
- 17. 気球野郎、大冒険.....50
- 18. スキーゲームPART1.....52
- 19. スキーゲームPART2.....54

●指先のテクニックが命!
絶妙コントロールゲーム

- 20. やぶから棒.....56
- 21. ブラックホールの掃除人.....58
- 22. パッシング・ロード.....60
- 23. 緊急入港ゲーム.....62
- 24. 万里の長城.....64
- 25. 横断大作戦.....66

●コチラと思えばまたアチラ!
反射神経養成ゲーム

- 26. 人間ハンティング.....68
- 27. モグラくんの1日.....70
- 28. ヘビ使い.....72
- 29. ベクトルゲーム.....74
- 30. 旗上げゲーム.....76

●ただひたすらタタケ!
単純興奮アクションゲーム

- 31. 100m走.....78
- 32. 押しずもう.....80
- 33. 自動車レース.....82
- 34. ダイヤモンド探し.....84

●あのナツカシのゲームももう一度!
古典的名作アクションゲーム

- 35. ブロック攻撃.....86
- 36. 平城京の闘い.....88
- 37. ブロック軍団の逆襲.....90
- 38. テニスゲーム.....92



大切なのは ゲームのストーリーだ




← 時に悪の味は旨い。

とはいっても、そう
考えこむことはない。




まずゲームのシチュエーション
を設定し……




宇宙が舞台、それとも海？


キャラクタのデザインと
動きを決めます。



①キャラクタは「鬼」と「人」



②カーソルキーで「鬼」を
動かし



たとえば「鬼ごっこ」という
ストーリーを考えました。



③「人」を捕まえたら
アナタの勝ち。



BASIC とはナニカ!?

パソコンは人間の言葉を
理解しません。

そこでパソコンにわかる
特別な言葉を使います。


そのひとつが
BASIC。人間の
言葉と同様、
単語や文法
があります。



たとえばPRINT文
という単語（コマンド）
を使うと……

画面に文字を書く
ことができます。

こんな言葉の組み合わせで
プログラムを作るのです。






- 100 画面の色などを決める
- 110 190行からデータを持ってくる
- 120 「人」にタッチしたら180行へ
- 130 キャラクターが現れる場所を決める
- 140 「人」の動きをコントロール
- 150 } 「鬼」の動きをコントロール
- 160 }
- 170 140行へ戻る
- 180 ゲーム終了
- 190 キャラクターのデータ
- ...

オニ コッコ

```

100 COLOR 15,4:SCREEN 1,1
110 GOSUB 190:SPRITE$(1)=S$:GOSUB 190:SPRITE$(2)=S$
120 ON SPRITE GOSUB 180:SPRITE ON
130 TIME=0:EX=60:EY=50:PX=200:PY=180
140 EX=RND(1)*20-10+EX:EY=RND(1)*20-10+EY:PUT SPRIT
E 0,(EX,EY),15,1
150 S=STICK(0):IF S=1 THEN PY=PY-6 ELSE IF S=3 THEN
FX=FX+6 ELSE IF S=5 THEN PY=PY+6 ELSE IF S=7 THEN
FX=FX-6
160 PUT SPRITE 1,(PX,PY),9,2
170 GOTO 140
180 PRINT "アッタタ !!":END
190 S$="":FOR I=1 TO 8:READ A$:S$=S$+CHR$(VAL("&H"+
A$)):NEXT:RETURN
200 DATA 1C,1C,4C,3C,0C,14,24,22
210 DATA 24,A5,7E,E7,A5,81,BD,7E
  
```

COLORというのは、背景色などを決める命令。



```
100 COLOR 15,4:.....
```

GOSUB 190.....190行に行って「人」と「鬼」のキャラクターデータ (S\$) を読んでこいという命令。



```
110 GOSUB 190:SPRITE$(1)=S$:.....
```

BASICの読み方!

行番号 プログラムの頭に順番に 100, 110... と並んだ数字が行番号。パソコンはこの行番号ごとに命令を実行する。

GOTO パソコンは行番号の若い順から命令を実行していくが、GO TO 文を使うと好きな行番号へ飛んだり、戻ったりできる。170 行の

GO TO 140 は、140 行へ戻れという意味なのだ。

GOSUB 110 行の GOSUB 190 は、190 行に行けという意味で GO TO 文と似ているが、190 行を実行した後、この行最後の RETURN 文でまた 110 行に戻る。GOSUB 文はブーメランのように元の行に戻る命令なのだ。なお、190 行のように何度でも使えるようまとめられた部

ON SPRITEというのは
「キャラクタ同士が重なったら」
てな意味。




120 ON SPRITE GOSUB ...

乱数を使って「人」をでた
らめに動かせといつて
ます。



140 EX=RND(1).....

「鬼」はカーソルで動かし
ます。




150 S=STICK(0)....

「鬼」が「人」にタッチしたら
こんな文字が現れます。

ヤツタネ!!

180 PRINT "ヤツタネ !!"



190 S#= " ":FOR

パソコンの描く絵は
小さな点(ドット)の
集まりです。
プログラム最後の
3行はこの点の位置
を決めているのです。


これでおしまい。
なーんともわかり
ましたか?



まとめデス

ストーリーを考える
シチュエーション、
キャラクタの形など

プログラムにする



パソコンに打ち込む



完成したプログラム
はテープに録音
(セーブ)



END,

分をサブルーチンと呼ぶ。

変数 130行や140行にEX, EYという記号
がでている。これらは計算した数値などを書き
こむメモのようなもので、変数と呼ばれる。た
とえば、EX=60はEXというメモに60という
数値を書いておけ、という意味。


IF~THEN~ELSE~ 150行にでてくるIF
文は「もし~なら」という判定をする命令。

条件があっていたら THEN に続く命令を実行し、
合っていなかったら ELSE に続く命令を実行す
る。

FOR~NEXT FOR と NEXT という言葉には
さまれた命令を決めた数だけ繰り返し実行させ
る。たとえば190行では変数Iが1から8にな
るまで、つまり8回分 READ... という命令を繰
り返すわけだ。

SPRITE機能

大図解




宇宙船からインベーダー、カワユイ女の子まで、アクションゲームの主演は面白キャラクターたち。このキャラクターを作ったり動かしたりするのがスプライト機能だ。アクションゲーム作りは、スプライト機能を使いこなすことから始めよう。

1☆SPRITEってナンだろう!?

スプライト機能は、どことなくマンガのアニメーションに似ている。アニメでは、背景となる絵の上に透明なセルに描いたキャラクターの絵を重ねる。あとは、セル画を少しずつ動かしながらフィルムに撮っていけば、いちいち背景を描かなくても、キャラクターが動いてみえるわけだ。


スプライト機能でもアニメと同じように、背景を描く画面とスプライト機能を使ってキャラクターなどを描く画面は分かれている。このスプライト画面(普通、スプライト面と呼ぶ)が、いわばアニメのセルに相当する。

さて、MSXパソコンには、このスプライト面が全部で32面用意されていて、それぞれの面に0~31の番号がついている。そして1つのスプライト面には、普通キャラクターを1個だけ描く。では、同時に何種類ものキャラクターが登場するゲームを作るときにはどうすればいいのだろうか? それぞれ異なったキャラクターを描いたスプライト面をいくつも重ね合わせればOKだ。




スプライト面は全部で32面。0から31までの番号で指定し、各面に1つキャラクターを表示。人間の眼には重なって見えるのだ。

2☆SPRITEの作り方



スプライト機能で作った絵は、実は8×8個、あるいは16×16個という小さな点の集まりなのだ。左の図が、その拡大図。さて、この点の集まりを横一列ごとにとりだし、黒い点を1、白い点を0とすれば、2進数の数字となる。さらにこの2進数を16進数(あるいは10進数)に直して、下のようにSPRITE\$(1)という配列に入れていけばできあがりだ。SPRITE\$(1)=CHR\$(&H18)+CHR\$(&H3C)+……
※&Hとは16進数のこと。

3☆SPRITEの使い方



PUT SPRITE(スプライト面番号),(X,Y),(色),(スプライト番号)

さて、実際にスプライト機能を使って絵を描くための命令がPUT SPRITE文。どのスプライト面に描くか、絵の色、座標、絵の種類を番号で指定していく。

まずスプライト面。前に注意したように、1つのスプライト面に描く絵は1つだけだから、同時に何種類も絵をだす場合は、それぞれ異なった面を使うこと。

座標、つまり絵を描く位置の値を少しずつ変えていけば、その絵はどんどん動いてみえる。このとき、色の番号を変えれば、絵が動くたびに、まるでネオンサインのように色が変化していく。また、同じ面に描く絵の種類を変えていくと、1つの絵がどんどん変身していきながら大ワザになるのだ。


ON SPRITE GOSUB(行番号)

人と車が衝突、あえなくゲームオーバー。こんなゲームに便利なのがON SPRITE文。2つのスプライト面に描かれた絵同士が、ちょっとでも触れたら、指定した行番号のサブルーチンを実行しろ、という意味。その行で「ドカンという衝突音を出せ」と命令してやればよい。



GRAPHIC 機能

大凶解




ゲーム作りに欠かせないのが背景。宇宙、化物屋敷、戦場などなど、背景が楽しければゲームもぐっと盛りあがる。背景や障害物を描くためのグラフィック機能は、オモシロゲームを作るためのポイントだ。

1★背景を描こう


LINE(X1, Y1) -
(X2, Y2), <カラー>
線を引くLINE文



LINE(X1, Y1) -
(X2, Y2), <カ
ラー>, B
長方形を描くLINE文



LINE(X1, Y1) -
(X2, Y2), <カラー>,
BF
長方形を塗りつぶす






CIRCLE<中心>,
<半径>
円を描く

PAINT(X, Y),
<カラー>, <境
界色>
境界色で囲まれた
部分を塗りつぶす



ゲームの操作方法


ゲームキャラクタは普通カーソルキーで操作することが多い。そんなとき使う命令が、S = STICK (0) だ。Sには右の絵のように、いま押されているカーソルキーの方向を示す値が代入される。上方向ならSの値は1、下と右方向を同時に押すと4となる。ちなみに押されていない場合は0だ。プログラムでは、この値に従ってキャラクタを移動していく。なお、STICK (0) の0を1に変えればカーソルキーの代わりにジョイスティックが使える。



2★障害物作りはゲームのポイント

障害物ゲームの基本

パックマンのように壁に沿ってキャラクターを動かす、爆弾に当たらぬように宇宙船をコントロールする……など、アクションゲームにとって、キャラクターの動きを制限する障害物は欠かせぬ要素だ。こんな障害物のあるゲームは、どんな構造になっているのだろう。たとえば、パックマンのようにせまい通路（障害物）の中でキャラクターを動かす場合を考えてみましょう。




まずカーソルキーで指示された方向（キャラクターの動く方向）に、壁があるか調べる。もし壁がなければキャラクターを動かし、壁があればキャラクターを動かさない、ということになる。つまり、障害物のあるゲームでは、常に障害物に突き当たっているかどうか調べながらキャラクターを動かしているというわけだ。


パソコンはこうして障害物を見分ける

パソコンの画面には横32×縦24文字（あるいは40×24文字分）を表示できる。そしてこれらの文字はすべてパソコンのメモリ内に記憶されており、画面上の表示される位置とメモリ内の記憶されている位置（番地という）は対応しているのだ。そこで、ある位置に障害物があるかどうか調べる場合、その位置に対応しているメモリ内の番地をみればよい。

画面とメモリ内の文字の関係を示すのが下の図だ。メモリの6144番地から1行（32文字分）ずつ24行が順番に並んでいる。たとえば第2行第3列の文字は $6144 + (2 \times 32 + 3)$ 番地に並んでいる。（ 40×24 文字表示の場合は、 $(2 \times 32 + 3)$ 番地）。あとは、 $C = VPEEK(6144 + (2 * 32 + 3))$ というプログラムで番地の文字がのぞける。そしてその文字が障害物かどうか比べてみればOKなのだ。



ゲームストーリー 発想法



講座



●キャラクタの行動パターンは4つ

① ATTACK型

敵を撃つ、破壊する



② CONTROL型

敵や障害物を避ける



③ GET型

モノを拾う、食べる




④ HIT型

飛んできたモノをはね返す



要素の中で特に重要なのがこの3つ。



4種類の行動パターンを合成する

CONTROL

HIT

CONTROL

+

+

+

GET


ATTACK

ATTACK

||

||

||



これでゲームの荒筋はできあがりです



あーは、はーは、はー

(バックマン)

(ブロック崩し)


(ギャラクシアン)

迷路を歩き (CONTROL)
エサを食べる (GET)


球を跳ね返し (HIT)
ブロックを壊す (ATTACK)

敵を避け (CONTROL)
撃つ (ATTACK)


●キャラクタデザイン




キャラクタの形を決める
だけではありません。




行動パターンに合わせ
細かい動きを決めます。




GET型のゲームなら何を拾うか、拾っ
たらどうなるか……




細部まできちんとデザインしたところで、
ゲームストーリーはほぼ完成。



ATTACK型なら、どんな武器を
使うか決めます。




●最後は背景



背景でゲームのイメージが
決まります、慎重に！




背景がただの飾りか、
障害物なのかも考えます。



パックマンも大変身

見なれたゲームも
背景次第で違ったゲーム
に見えます



森の中の大冒険

妖しいピラミッド

面白くなるゲーム



4 文字をだす

ゲーム終了時に画面に得点が表示される、ハイスコアをだした人の名前を登録できる……こんな工夫があれば、プレイヤーのヤル気も違います。




5 難易度を 変える

初心者からゲームに慣れたマニアまで、だれでも楽しめるように、ゲームのスピードや敵の数をプレイヤー自身が自由に選べるようにします。



6 背景を 変える

一画面を突破するたびにゲームの背景を街から海へ、海から宇宙へといった具合に変えていきます。単純な動きのゲームに興行性が生まれます。



星をワープさせるために、125行でON INTERVAL AL……GOSUB 185という命令が書かれている。一定時間ごとに指定した行(185行)へ行けという命令だ。そして185行の命令で星はワープする。

180行では、宇宙船が星にタッチするたびにエネルギーを補給する(E = E + 10)ことになっている。

そして、最後に115行で宇宙を思わせる背景を描いたらプログラム完成だ。

```

100 COLOR 15,4:SCREEN 1,1
110 GOSUB 190:SPRITE$(1)=S$:GOSUB 190:SPRITE$(2)=S$
115 FOR I=0 TO 100:LOCATE RND(1)*29,RND(1)*24:PRINT
".":NEXT
120 ON SPRITE GOSUB 180:SPRITE ON
125 ON INTERVAL=100 GOSUB 185:INTERVAL ON
130 TIME=0:EX=60:EY=50:FX=200:FY=180
135 E=200:T=0
140 EX=RND(1)*20-10+EX:EY=RND(1)*20-10+EY:PUT SPRIT
E 0,(EX,EY),9,1
150 S=STICK(0):IF S=1 THEN FY=FY-8 ELSE IF S=3 THEN
FX=FX+8 ELSE IF S=5 THEN FY=FY+8 ELSE IF S=7 THEN
PX=PX-8
160 PUT SPRITE 1,(PX,FY),15,2
165 T=T+1:E=E-1:LOCATE 0,0:PRINT "TIME ";T;"ENRGY "
;E:IF E=0 THEN PLAY "04L32GFEDC":END
170 GOTO 140
180 SPRITE OFF:BEEP:E=E+10:SPRITE ON:RETURN
185 EX=248*RND(1):EY=184*RND(1):RETURN
190 S$="":FOR I=1 TO 8:READ A$:S$=S$+CHR$(VAL("&H"+
A$)):NEXT:RETURN
200 DATA 00,0C,0E,0F,07,0D,03,07
210 DATA 18,24,24,7E,FF,FF,7E,00
    
```


覚えておくと便利だよ!

プログラムテクニック1ポイントレッスン

☆関数を利用しよう

たとえば $y = a \times x + b$ のように、値 x を変えるとそれにつれて値 y も変わっていくような式を関数という。こうした関数は `DEF FN <関数名> = VAL (MID $("101112111", S + 1, 1)) - 1` といった具合に一度定義しておけば、計算のたびに式を書かなくても、関数名だけ書けばいいので便利。

☆パソコン内の時計を利用しよう

パソコンの時計代わりになるのが TIME。1/60秒ごとにひとつずつ値が増えるので、ちょっと計算すれば時間がわかる。また一定時間ごとにスコアを表示したり、キャラクタを移動させたりする場合、`ON INTERVAL = <時間間隔> GOSUB <行番号>` が便利。時間がくれば <行番号> の命令を実行してくれるのだ。

☆効果音の出し方

BASIC で音をだす命令が PLAY と SOUND。PLAY はだしたい音を文字で指定していく。たとえば PLAY "A" なら「ラ」の音がでる。詳しくは MSX のマニュアルを読んでほしい。SOUND は、爆発音や機械音など、細かい効果音をだすとき使う。もう一つ、ピーツという音をだす BEEP も覚えておこう。

☆ミサイル発射にスペースバーを使おう

キャラクタはカーソルキーで操作する。ではミサイルやビーム砲の発射スイッチはというと、スペースバーを使うのが一般的。STRIG (0) は、スペースバーが押されると値が -1、押されないと 0 になる関数。これを利用すれば、いまスペースバーが押されているかどうか調べることができる。さらに便利なのが `ON STRIG GOSUB <行番号>`。スペースバーが押されたら指定した行番号に行けという命令だ。そこにミサイルを発射するプログラムを書いておけば、スペースバーが発射スイッチになるのである。

☆その他の主な命令

`RND (1)`, `RND (-TIME)`……`RND (1)` は 0 から 1 未満のランダムな数を作るときに使う。そして `RND (-TIME)` を使うとゲームのたびに異なったパターンの乱数となる。

`KEY OFF`……画面下にあるファンクションキーの表示を消す。

`DEFINT A-Z`……すべての変数を整数にする命令。これで計算が単純になる、つまり速くなるのだ。

`READ, DATA`……プログラムの中にある DATA を読むのが READ。ゲームキャラクタのパターンなどを読むのによく使われる。

ACTION GAME

38

●硬くなった脳ミソをほぐそう！思考型アクションゲーム

アクションゲームといっても、バクюн、バクюнのアタマイランゲームばかりじゃないぞ。たまには頭も酷使するアクションゲームで苦しんでみよう。

●欲求不満のあなたに！爽快シューティングゲーム

ナンカコー、すっきりしない、もやもやする、イライラするの3拍子がそろったら、危険信号。さあ、気分をかえてシューティングゲームでスッキリ。

●画面が動けば心もおどる！スクロールゲーム

ゲームにも臨場感が欲しいな、というキミにはスクロールゲームだ。画面が流れるスクロールゲームなら臨場感はヒカイチだ。

●指先のテクニックが命！絶妙コントロールゲーム

勉強でもスポーツでもテクニックが大事。ゲームだって同じこと。キミの指先の動きが勝負を決定する、スリリングな絶妙コントロールゲームでキマリ。

●コチラと思えばまたアチラ！反射神経養成ゲーム

パソコンの前にばかり座っていると、運動不足になりかねないゾ。治療はやはりパソコンゲームで。反射神経がメキメキつく反射神経養成ゲームが最適だ。



●ただひたすらタタケ！単純興奮アクションゲーム

ムズカシイことは考えたくない、1日中ボーッとしたい人はゲームなんかしないで寝ていればいい？ いやいや、単純興奮ゲームこそおすすめしたい。

●あのナツカシのゲームよもう一度！古典的名作アクションゲーム

あの頃はよかった、あの頃のゲームはよかった——というわけで懐古調ゲームはいかがかな？ 古典と呼ばれるゲームはいまでも不思議と新鮮だ。

＜入力時の諸注意＞

- 1). プログラムを1行打ち終わることに、 (リターンキー)を押すこと。2). プログラム中の", ; , : , ; , "などの記号を忘れずに打つこと。3). 数字のゼロ(0)と英数字のオー(O)を間違えないこと。
- 4). プログラムを入力し終わったら、RUNと打ちこんで (リターンキー)を押すか、ファンクションキーの[F5]を押してください。

1

放火はイケナイ！でも楽しい 放火魔

すべての大学の受験に失敗し、二浪が確定的になったボクは、むしゃくしゃした気分で夜中の散歩に出た。家の窓からはいかにも家庭的な明るい笑い声が聞こえてくる。ボクは思わずムカッときて、気がつくと1軒の家に火をつけようとしていた。



PLAY


キー操作が変わっている。左右のカーソルキーだけで、画面にいっぱいある家々の間に、放火魔を走らせるのだ。どうなっているのかというと、左右の両端どちらにでも出ると、その下の段の反対側から現れるようになってきているのだ。いちばん下へいったらつぎはいちばん上に出る。

みんなも知っている(?)だろうが、家に確実に火をつけるには時間がかかる。このゲームでも、各家の真下でしばらく待たねばならない。ところがぐずぐずしているとパトカーが近づいてくる。最初のほうの面では簡単に逃げ切れるのだが、面が進むとだんだんパトカーが速くなり、むずかしくもおもしろくなるのだ。



PROGRAM

パトカーは、四つ角にくるたびに方向を転じて放火魔に近づいてくる。四つ角にきたかどうかを判断しているのが340行だ。放火魔の動きは300行。一見複雑そうだが、プログラムは実はシンプル。複雑に見える動きを簡単なプログラムで処理している。

ちなみに、パトカーが上下移動するとき、一瞬画面から消えるときがあるが、これはスプライトが同一走査線上に4つしか表示できないのに、家4軒とパトカー、合計5つ存在したために、パトカーが表示されなくなるというわけだ。



再プレイ……RUN (F・5)



また1軒全焼。罪悪感をかみしめておつぎへ

思考型アクションゲーム

```

100 '*** houka ***
110 '
120 COLOR 15,1:SCREEN 2.2:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DEFFNR(A)=RND(1)*A
140 DEF FNX=VAL(MID$("111211101",STICK(0)+1,1))-1:OPEN "
GRP:" AS #1
150 FOR I=1 TO 4:S$=""
160   FOR J=1 TO 32
170     READ A$:S$=S$+CHR$(VAL("&H"+A$))
180   NEXT J
190   SPRITE$(I)=S$
200 NEXT I
210 ON SPRITE GOSUB 380:SPRITE ON
220 '
230 CO=0:LE=LE+1:SC=SC+500:PX=184:PY=176:CX=56:CY=32:AD=
2*(LE¥3)MOD4+2)
240 CLS:FOR J=0 TO 3:FOR I=0 TO 3
250   PUT SPRITE 1+J*4,(24+64*I.16+48*J),2,3:HJ(I,J)=0
260 NEXT I,J
270 PRESET (16,0):PRINT #1,"SCORE"
280 '
290 LINE (72,0)-(108,8),1,BF:PRESET(72,0):PRINT #1,USING
"#####":SC
300 PX=PX+FNX*8:IF PX>232 OR PX<0 THEN PY=(PY+48)MOD192:
PX=(PX+240)MOD240
310 PUT SPRITE 20,(PX,PY),15,2
320 IF PX MOD 64<>24 THEN PJ=9 ELSE PJ=PJ-1:A=PX¥64:B=PY
¥48:IF PJ<1 AND HJ(A,B)=0 THEN HJ(A,B)=1:SC=SC+20:PLAY "
L64N40":PUT SPRITE A+B*4,(PX,PY-16),9,4:CO=CO+1:IF CO=16
THEN PLAY "FECEFAB":GOTO 230
330 IF ABS(CYMOD48-32)<AD THEN CY=(CY¥48)*48+32
340 IF ABS((CX-56)MOD64)<7 THEN AX=SGN(PX-CX):AY=SGN(PY-
CY):IF AY<>0 THEN AX=0
350 CX=((CX+16)+AX*AD+288)MOD288-16:CY=CY+AY*AD:PUT SPRI
TE 30,(CX,CY),15,1
360 GOTO 290
370 '
380 PLAY "L64BAGFEDL4C":SCREEN 1:PRINT "LEVEL";LE,"SCORE
":SC
390 END
400 '
410 DATA 00.00.00.00.01.01.07.08.10.7F.80.80.9C.E3.22.1C
420 DATA 00.00.00.00.80.80.E0.10.08.FE.01.01.1D.E3.22.1C
430 DATA 03.03.03.01.03.07.0F.1B.33.23.03.03.06.06.06.07
440 DATA C0.C0.C0.80.C0.E0.F0.D8.CC.C0.C0.60.30.30.30.38
450 DATA 01.03.06.0C.18.30.60.EF.28.28.28.28.2F.20.20.3F
460 DATA 80.C0.60.30.18.0C.06.87.84.BC.A4.A4.B4.24.24.FC
470 DATA 04.04.0E.2E.3F.7F.7E.6E.44.44.44.44.20.20.20.1F
480 DATA 40.40.E4.E4.FE.FE.FE.66.66.46.42.02.04.04.04.F8

```


2

南海の流れ島「ひょうたん島」をめざせ!

スカイダイビング

ゲームソフトで巨万の富を築いたキミは、南海の島「ひょうたん島」を買った。しかし、ナントこの島は流れ島だったのだ! 島に着地する手段は、飛行機からのスカイダイビングのみ。さて、キミは着地に成功し、優雅な孤島生活ができるかな?



PLAY

RUN すると画面の左側に「ひょうたん島」上空の風の情報が表示される。風は「<」なら左向き、「>」なら右向きである。風の強さは、風向き記号の数で表される。サテ、飛行機が飛んできたゾ! 画面右下からは……「ひょうたん島」がゆっくり登場。




風の向きは? 強さは? よしいまだ、飛び降りるゾ! ポンと軽くスペースバーを押す。風に吹かれてパラシュートが、右に左にゆらりゆらりと降りていく……アー、「ひょうたん島」から離れるウ。よし、パラシュートを操作して風に対抗だ。カーソルキーの左右で、風の効果を加減できるのだ。

し、しかし、努力のカイなくスカイダイバーは海にドボン。ムム、あと2人しか残っていない、つぎは必ず成功するゾ。慎重に、慎重に……。

PROGRAM

このゲームで重要な役割を担っているのが風。この風のデータは、W(1)という配列に入っている。W(1)は170行で $W(1) = \text{FNR}(10) - 5$ と $+5 \sim -5$ の値をセットされている。W(1)が負なら左向き、正なら右向きの風を、そして絶対値が風速を示すのだ。

このW(1)をもとに、風の情報を画面上に「<」、「>」で示し、パラシュートを動かす(280行)。ちなみに220~290行の **FOR~NEXT** 文で、上からの風の影響を順次加えていくのである。



オット! 風に流されはじめたぞおい!

思考型アクションゲーム

```

100 '*** skydiv ***
110 '
120 COLOR 1,7:SCREEN 1,2:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DIM W(24):DEFFNR(X)=RND(1)*X
140 GOSUB 400:LV=1:PL=3:ON SPRITE GOSUB 330:ON STRIG GOS
UB 350
150 GOSUB 370
160 FOR I=3 TO 20
170   W(I)=FNR(10)-5:LOCATE 0,I
180   PRINT STRING$(ABS(W(I)),CHR$(61+SGN(W(I))))
190 NEXT:F=0:X=0
200 X=(X+4) MOD 250:PUT SPRITE 1,(X,0),14,1
210 IF F=0 THEN FOR W=0 TO 200:NEXT:GOTO 200
220 FOR Y=3 TO 23:FOR I=0 TO ABS(W(Y))
230   PUT SPRITE 0,(PX,Y*8-8),15,0
240   FOR W=0 TO 50
250     S=STICK(0)
260     NEXT
270     IF S=3 OR S=7 THEN PX=PX-(S-5)*2
280     PX=PX+SGN(W(Y))*8
290 NEXT:NEXT
300 PUT SPRITE 0,(0,209),0,0:PL=PL-1:LOCATE 25,0:PRINT U
SING "#":PL:PLAY "L32GFEDC"
310 IF PL=0 THEN END ELSE F=0:STRIG(0) ON:GOTO 200
320 '
330 LV=LV+1:PUT SPRITE 0,(0,209),0,0:GOTO 150
340 '
350 F=1:PX=X:STRIG(0) OFF:RETURN
360 IS=(IS+248)MOD 256:PUT SPRITE 2,(IS,180),2,2:PLAY "L
802A":RETURN
370 PLAY "S8M100004DBG":CLS:LOCATE 12,0:PRINT USING "PAG
E ## LEFT #":LV:PL
380 ON INTERVAL=500/(LV+4) GOSUB 360:SPRITE ON:STRIG(0)
ON:INTERVAL ON:RETURN
390 '
400 FOR J=0 TO 2
410   S$=""
420   FOR I=1 TO 32
430     READ A$:S$=S$+CHR$(VAL("&H"+A$))
440   NEXT
450   SPRITE$(J)=S$
460 NEXT
470 RETURN
480 DATA 03,0D,3E,77,6D,FF,D5,80,41,31,0F,05,03,02,02,06
490 DATA C0,B0,7C,EE,B6,FF,AB,01,82,8C,F0,A0,C0,40,40,60
500 DATA 00,00,00,03,C4,E4,E4,E4,FC,FC,FF,40,40,00,00,00
510 DATA 00,00,00,00,80,80,99,A5,FF,FF,FD,21,61,60,00,00
520 DATA 00,00,0E,1F,7F,7F,FF,FF,FF,FF,00,00,00,00,00,00
530 DATA 00,00,00,7C,FE,FE,FF,FF,FF,FF,00,00,00,00,00,00

```


3 パトカーを出し抜いて銀行を襲え!

バンク・パニック

ニューヨーク，ウォール街。午前2時になるとここはギャングと警察が活躍する舞台となる。警察はパトカーで巡回し，怪しい行動をする者を片っ端からつかまえ，ギャングは警察につかまらぬよう物陰を利用して銀行を襲う。生死をかけた闘いだ。



PLAY




整然と碁盤の目のように走る道路網の中に，25の銀行が立ち並んでいる。銀行には1～25の番号がついているが，並ぶ順序はバラバラだ。ギャングであるオレは警察につかまることなく番号順に襲撃しなければならない，しかも制限時間内に。

パトカーが自分と同じ南北，または東西の通りにいて，しかも，ある程度近づいたらそれだけでつかまってしまう。だから銀行の中か入口に身を潜ませながら行動しなければ，それに，番号をまちがえて侵入するとポイントダウンになるから，要注意だ。

PROGRAM

まず，銀行をランダムに並べ換える方法を見てみよう。210行に注目。最初のFOR～NEXT文で，銀行に順番に“通し番号”をつける。つぎのFOR～NEXT文で25個の中からランダムに2つを選び，その番号を入れかえる。これを何度も繰り返せばほとんどランダムに並ぶというわけ。

また，パトカーがギャングを見つけたかどうかの判定法は，400行を見てくれ。南北の道路ではパトカーとギャングのX座標の一致でわかる。だが東西の道路ではギャングが建物の反対側にいることがあるので，Y座標が東西の道路上にあるかどうかの判定も必要なのだ($Y \text{ MOD } 4 = 0$)。



レレ……？ つぎの番号はどこだ!?

思考型アクションゲーム

```

100 '*** p-gang ***
110 '
120 COLOR 15,1:SCREEN 1,0:WIDTH 32:KEY OFF:DEFINT A-Z:R=
RND(-TIME):DIM B(25)
130 DEFFNR(X)=RND(1)*X+1:DEFFNP(X,Y)=VPEEK(Y*32+X+6144)
140 DEFFNY(S)=VAL(MID$("101112111",S+1,1))-1:DEFFNX(S)=V
AL(MID$("111211101",S+1,1))-1
150 PLAY "T100L8"
160 FOR I=0 TO 1:S$=""
170   FOR J=1 TO 8:READ A$:S$=S$+CHR$(VAL("&H"+A$)):NEXT
   J
180   SPRITE$(I)=S$
190 NEXT I
200 PN=3:LOCATE 14,0:PRINT USING "Left #":PN
210 FOR I=1 TO 25:B(I)=I:NEXT I
220 FOR I=0 TO 100
230   SWAP B(FNR(25)),B(FNR(25))
240 NEXT I
250 FOR I=0 TO 4:FOR J=0 TO 4
260   C=J*6+2:L=I*4+3:LOCATE C,L:PRINT " ┌───┐ "
270   LOCATE C,L+1:PRINT " |   | "
280   LOCATE C,L+2:PRINT " └───┘ ";
290   LOCATE C+1,L+1:PRINT USING "##":B(I*5+J+1)
300 NEXT J,I:BC=1
310 TIME=0:X=23:Y=22:ON INTERVAL=40 GOSUB 370:INTERVAL 0
   N
320 FOR W=0 TO 20:NEXT W:S=STICK(0):DY=FNY(S):DX=FNX(S):
X=X+DX:Y=Y+DY
330 IF FNP(X,Y)<>32 OR X<1 OR X>31 OR Y<2 OR Y>22 THEN X
=X-DX:Y=Y-DY
340 PUT SPRITE 0,(X*8,Y*8),15,0
350 IF (X MOD 6)=5 AND (Y MOD 4)=0 THEN 360 ELSE 320
360 IF B(X*6+(Y*4-1)*5+1)<>BC THEN PLAY "L6403CL8":P=P-5
:GOTO 320 ELSE PLAY "L6404AR64AL8":P=P+100:BC=BC+1:LOCAT
E X,Y:PRINT "*":IF BC=26 THEN INTERVAL OFF:GOTO 210 ELSE
   320
370 LOCATE 0,0:PRINT USING "Score ##### ";P:LOCATE 21,0:
PRINT USING "TIME #####":300-TIME/60
380 IF TIME>18000 THEN GOSUB 410:F=0:TIME=0:GOTO 310 ELS
E IF F=1 THEN 390 ELSE F=1:PX=FNR(6)*6-5:PY=2:RETURN
390 PLAY "O5GA":PUT SPRITE 1,(PX*8,PY*8),7,1
400 IF ((PY=Y AND (Y MOD 4)=2) OR PX=X) AND (X-PX)^2+(Y-
PY)^2<197 THEN GOSUB 410:F=0:RETURN 310 ELSE PY=PY+2:F=S
GN(PY MOD 24):RETURN
410 PLAY "L6402BAGFEDCL8":PN=PN-1:LOCATE 14,0:PRINT USIN
G "Left #":PN
420 IF PN=0 THEN END ELSE PY=2:RETURN
430 DATA 1E,3F,16,1E,0C,EE,3E,2E
440 DATA 00,10,7C,44,FF,FF,66,00

```


4 ファンコロガシの気持ちがわかる 酒樽ころりん

地震だ！ 酒屋の倉庫は大混乱に陥った。荷くずれした酒樽がつぎからつぎへと転がり落ちてくるのだ。キミはたったひとりでのこの混乱を治めなければならない。樽を1個ずつ押して転がして行くのだから重労働だぞ。



PROGRAM

酒樽はどんどん転がってくる。画面は4段の層に分かれていて、樽は右から左へ転がり、左端へ達するたびにその下の段のまた右端から転がってくるのだ。で、その樽を1個ずつ、転がってきた方向と逆に、つまり右に押していく。右端に達するたびにその上の段の左端に出られるので、がんばって最上段の右端に収めねばならない。転がり落ちる樽をつかまえてそこねて、最下段の左端までいかせたら、負け。




ところが意外にむずかしいのは、樽は1個ずつ転がり落ちてくるわけではないという点。こっちを押していたら下の段でも転がっていく樽が、やばい！ あっちが先だ！ 自分のキャラクターはカーソルキーで左右だけでなく上下にも飛び移れるのだが、つぎつぎと現れる樽を運ぶのは、実に忙しいのだ。

PLAY

ある1つの動きを処理する方法は1つとは限らないので、よりよい処理法を見つけねばならない。

たとえば酒樽を移動させるとき、1個ずつについて位置やぶつかったかどうかを調べていたのでは、すばやさが全然なくて反射ゲームにならない。

そこでこのプログラムでは、道の上を PD\$ という文字変数に入れて、PD\$ を MID\$ 命令で区切って表示させた。酒樽の移動は320行の処理で OK、ぶつかったかどうかは460~480行で処理できる。



上を転がせば下が転がる、と不条理な世界だぞ

思考型アクションゲーム

```

100 * *** roll ***
110 *
120 CLEAR 1000:COLOR 15.1:SCREEN 1.0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEFFNA(A,B)=(3-B*5)*29+1+A:DEFFNR(A)=RND(1)*A:DEFFNM
$(B)=MID$(PD$,B,1)
150 DEFFNX=VAL(MID$("112221000",STICK(0)+1,1))-1
160 DEFFNY=VAL(MID$("100122210",STICK(0)+1,1))-1
170 FOR I=1 TO 8
180   READ A$:S$=S$+CHR$(VAL("&H"+A$))
190 NEXT I
200 SPRITE$(1)=S$
210 *
220 PD$=SPACE$(116):LE=LE+1
230 FOR I=0 TO LE
240   X=FNR(29):Y=FNR(2)*5+4:MID$(PD$,FNA(X,Y),1)="●"
250 NEXT I
260 FOR I=1 TO 4
270   LOCATE 0.5*I:PRINT STRING$(29,"□")
280 NEXT I
290 PX=0:PY=19
300 LOCATE 15,0:PRINT "LEVEL":LE
310 *
320 PD$=MID$(PD$,2)+" ":PLAY "L1604CA":GOSUB 460
330 FOR ZC=0 TO LE
340   PX=PX+FNX:PY=PY+FNY*5:IF PX<0 THEN PX=28:PY=PY+5 E
LSE IF PX>28 THEN PX=0:PY=PY-5
350   IF PY<4 THEN PY=4
360   IF PY>19 THEN PY=19
370   GOSUB 460:PUT SPRITE 0,(PX*8+16,PY*8),7,1
380   FOR I=0 TO 3
390     LOCATE 0.19-I*5:PRINT MID$(PD$,FNA(0.19-I*5),29)
400   NEXT I
410 NEXT ZC
420 IF PD$=SPACE$(116) THEN FOR I=1 TO 3:PLAY "L6404CDEF
G":NEXT I:FOR W=0 TO 2000:NEXT W:GOTO 220
430 IF LEFT$(PD$,1)="●" THEN PLAY "L203A":END
440 GOTO 320
450 *
460 SL=FNA(PX,PY):IF FNM$(SL)<>"●" THEN RETURN
470 IF SL=115 THEN MID$(PD$,SL,2)=" ":PLAY "L6406ABAB":
RETURN
480 IF FNM$(SL+1)="●" THEN IF PY=19 THEN PLAY "R16L6404G
FEDL2C":END ELSE MID$(PD$,SL-29)="●":PLAY "R16L.6404GFEDC
"
490 MID$(PD$,SL,2)=" ●":RETURN
500 *
510 DATA 38,38,17,38,3F,38,CC,86

```


5

寄ってくるやつは嫌い、逃げるやつは好き

磁石の惑星

人類は宇宙へ飛び出し、さまざまな天体を開拓しようとしていたっていうから未来の話だ。そして、ある開拓団が着陸したのがこの惑星。ところが強力な磁気を帯びた物質が無数に落ちている。からだにくっつくやつもあって命とりだ。どうする!?



PLAY

画面を見ると左側に川がある。川の右側にちらばっている“-”は磁石だ。画面にはキミが操作する人間もいるが、その人間に近づいてくる磁石と反発して離れていく磁石がある。キミは反発する磁石に接近して、反発力で川に落とさねばならない。制限時間内に5個落とせばつぎの面に進める。ところが近づいてくる磁石がくせもので、そいつの磁界に踏みこんで自分のほうに動かしたらそれだけで減点だし、からだに4個くっつけたらゲームオーバーなのだ!




それではほとんど見分けのつかない2種類の磁石をどう見分けたらよいのだろう。実はよく見ると長さが違うのだ。短いほうが反発する。

面が変わると近づく磁石が増えていやすくなる。

PROGRAM

磁石は、主人公自身から見て、正面と真横にきたときに接近していれば反応する。ところが、進行方向に対する相対的な正面、真横なわけだから、主人公が動くたびに方向がどう変わったか調べねばならない。この方向決定が300行の $X(1)=\sim$ と、 $Y(1)=\sim$ で行われる。

さて、このプログラムでは、主人公と磁石の距離が3以内だと反応するようになっている。そこで、反応距離を変えたい人には310行の FOR 文に注目していただきたい。そこの数値を変えればいいのだ。



それ! この短い磁石を川へはじき飛ばせ!

思考型アクションゲーム

```

100 '*** jishaku ***
110 '
120 COLOR 15,1:SCREEN 1,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DEF FNR(A)=RND(1)*A
140 DEF FNX(S)=VAL(MID$("112221000",S+1,1))-1
150 DEF FNY(S)=VAL(MID$("100122210",S+1,1))-1
160 DEF FNA(X,Y)=6146+X+Y*32
170 DEF FNP(X,Y)=VPEEK(FNA(X,Y))
180 FOR I=1 TO 8:READ A$:S$=S$+CHR$(VAL("&H"+A$)):NEXT:S
PRITE$(1)=S$
190 '
200 LE=LE+1:PX=27:PY=10:PC=0:FC=5:CLS
210 ON INTERVAL=6040 GOSUB 430:INTERVAL ON:TIME=0
220 FOR I=2 TO 21:LOCATE 0,I:PRINT STRING$(4,"~"):NEXT I
230 FOR I=0 TO LE*10:LOCATE FNR(21)+5,FNR(20)+2:PRINT CHR$(176):NEXT I
240 FOR I=1 TO 12:LOCATE FNR(21)+5,FNR(11)+5:PRINT CHR$(45):NEXT I
250 '
260 A=STICK(0):IF A>0 THEN S=A
270 PX=PX+FNX(S):PY=PY+FNY(S):IF ABS(PX*2-33)>24 OR ABS(PY*2-23)>20 THEN PX=PX-FNX(S):PY=PY-FNY(S)
280 PUT SPRITE 1,(PX*8+16,PY*8),15,1
290 FOR I=0 TO 2
300   X(I)=FNX((S+I*2+6)MOD8):Y(I)=FNY((S+I*2+6)MOD8)
310   FOR J=0 TO 3
320     X=X+X(I)*J:Y=PY+Y(I)*J:P=FNP(X,Y)
330     IF P=45 THEN VPOKE FNA(X,Y),32:X=X+X(I)*3:Y=Y+Y(I)*3:IF X<4 THEN PLAY "L6407CDEFG":FC=FC-1:PO=PO+100 ELSE
GOSUB 410:PLAY "L6406CDE":VPOKE FNA(X,Y),45
340     IF P=176 THEN VPOKE FNA(X,Y),32:IF J<2 THEN PC=P
C+1:PO=PO-50:PLAY "L6403GFEDC" ELSE VPOKE FNA(X-X(I),Y-Y(I)),176:PO=PO-10:PLAY "L6403DC"
350   NEXT J
360 NEXT I
370 IF PC>3 THEN 430
380 IF FC<1 THEN A=100-TIME¥60:FOR I=1 TO A:LOCATE 8,8:PRINT USING "BONUS #####":(A-I)*10:PO=PO+10:GOSUB 400:PLAY "L64N50":FOR W=0 TO 30:NEXT W,I:FOR W=0 TO 3000:NEXT:GOTO 200
390 LOCATE 0,0:PRINT "HAVE":STRING$(PC,"-"):TAB(8):"LEFT":FC:LOCATE 16,1:PRINT USING "TIME ###":100-TIME¥60:GOSUB 400:GOTO 260
400 LOCATE 16,0:PRINT USING "SCORE #####":PO:RETURN
410 IF X>28 THEN X=28
420 RETURN
430 PLAY "L6404BGECL403C":END
440 DATA 1C,DC,68.3C,FC,1C,36.33

```


6

怪物戦車に歩兵がひとり立ちむかう！ 手榴弾を投げる！！

第24歩兵部隊の前に突如として出現した、敵の巨大な重戦車。たちまち戦闘が始まった。が、さしもの勇猛な歩兵部隊も戦車の敵ではなかった。残るは、あなたひとり、しかも武器は手榴弾のみなのだ。さあ！？



PLAY




手榴弾の射程距離は5だが、カーソルキーで人間が動いている方向にしか投げられない。これに対し、戦車の射程距離は3と短い、そのなかに人間がいると必ずやられてしまう。戦車をやっつけると、残り時間がボーナス点になるので、急いで行動する必要はあるのだが、不用意に近づくと危険、危険！

戦車も歩兵も、爆弾によってできた穴を越えられないから、手榴弾で穴をつくって防御線をしくのが作戦としてはよいのでは。

PROGRAM

敵の動きをどうするかは、ゲームのおもしろさを左右する重大要素。敵の動き方としては、とりあえずつぎの3種が考えられる。まず、まったくランダムな動き、つぎに、プレイヤー側のキャラクターに近づいてくる動き、そして、この2つをミックスした動きだ。

このゲームでは、第3の、ランダムな動きと近づく動きをミックスさせる方法をとっている。そのときどきで、どちらの動きを選ぶかは、290行のIF文で決めている。そして、ランダムな動きは310行で行っている。近づく動きは、まず関数FNA(A, B)を150行で決め、それを使って290行で行っているというしかけだ。



ねらいがはずれて敵戦車のそばに穴があく

爽快シューティングゲーム

```


100 ' *** tank ***
110 '
120 COLOR 15.1:SCREEN 1:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEFFNX(S)=VAL(MID$("112221000",S+1,1))-1:DEFFNY(S)=V
AL(MID$("100122210",S+1,1))-1
150 DEFFNR(A)=RND(1)*A:DEFFNA(A,B)=A+SGN(B-A):DEFFNP(X,Y
)=VPEEK(6146+32*Y+X)
160 '
170 LE=LE+1:TT=100+300/LE:PO=PO+100:CX=2:CY=2:PX=26:PY=2
1
180 LOCATE 1,1:PRINT STRING$(27,"□"):FOR I=2 TO 21:PRINT
" □":SPC(25);"□":NEXT I:PRINT " ";STRING$(27,"□")
190 FOR I=0 TO 10
200   LOCATE RND(1)*21+4,RND(1)*16+4:PRINT "○"
210 NEXT I
220 '
230 LOCATE PX,PY:PRINT " ":S=STICK(0):BX=FNX(S):BY=FNY(S
):X=PX+BX:Y=PY+BY:GOSUB 370
240 IF FNP(X,Y)=32 THEN PX=X:PY=Y
250 LOCATE PX,PY:PRINT "◆"
260 IF STRIG(0)=0 OR S=0 THEN 290
270 X=PX+BX*5:Y=PY+BY*5:PLAY "T255L64N80":GOSUB 370
280 LOCATE X,Y:PRINT "○":IF X=CX AND Y=CY THEN LOCATE 10
,10:PRINT "ヤッタネ !":PLAY "L6405CEGA":FOR I=1 TO TT:TT=TT-
1:PO=PO+1:GOSUB 410:NEXT I:GOTO 170
290 X=FNA(CX,PX):Y=FNA(CY,PY):IF FNR(8/(LE*.3+1))=0 THEN
320
300 IF FNR(8/(LE*.3+1))>0 THEN X=CX+FNR(3)-1:Y=CY+FNR(3
)-1
310 X=CX+FNR(3)-1:Y=CY+FNR(3)-1
320 GOSUB 370:IF FNP(X,Y)=32 THEN LOCATE CX,CY:PRINT " "
:CX=X:CY=Y ELSE 310
330 TT=TT-1:PO=PO-1:GOSUB 410:LOCATE CX,CY:PRINT "♣"
340 IF (CX-PX)^2+(CY-PY)^2<10 THEN LOCATE PX,PY:PRINT "X
":GOTO 440
350 IF TT<1 THEN 430 ELSE 230
360 '
370 IF X<2 THEN X=2 ELSE IF X>26 THEN X=26
380 IF Y<2 THEN Y=2 ELSE IF Y>21 THEN Y=21
390 RETURN
400 '
410 LOCATE 1,0:PRINT USING "TIME ####  ":TT::PRINT USING
"SCORE #####":PO:RETURN
420 '
430 LOCATE 0,5:PRINT "TIME OVER"
440 PLAY "L6404GFEDC":END

```


2

ふたりで戦う戦車戦だあ!!

タンク-2



こいつは、いままでのゲームとチョイとちがう。ふたりで戦う壮絶な戦車戦だ！ ひとりでパソコンと遊ぶのも楽しいけど、ここはひとつ、友だちや兄弟に戦いを挑んで、ギャフンといわせてやろう。なにしろキミ自身が制作したゲームなんだからネ。

PLAY


この血沸き肉踊る(ホント!)戦車戦をやるには、ジョイスティックが1本必要です。ひとりがキーボード上のカーソル移動キーで、左上に現れた戦車を操作する。スペースバーをたたけばミサイル発射だ。右下の戦車を操作するのは、ジョイスティックを扱う人だ。うまく動いて敵戦車を撃破!

早く5勝したほうの勝ちである。十二分に障害物を利用して、お互いのかけ引きの中で相手をうまくダメした方に勝利の女神は微笑むであろう。おーい、誰か勝負しよーよー。


PROGRAM

ふたり用だからといって、ひとりめとふたりめの処理を別べつにしているわけではない。240行から320行を見てほしい。「STICK」と「STRIG」は、カッコのなかの数値を変えることにより、キーボードにもジョイスティックにも対応するし、ミサイルや戦車の座標なども、配列に入れるとまったく同一の処理となり、ひとつに統一できる。

なお、よく使う処理はサブルーチンにしておくときよい(370行, 380行, 390行)。別のゲームをつくったときに、同じ処理を行なう部分がでてきた場合、そのサブルーチンをそっくりそのまま使用することができるからである。



プレイヤー2……ジョイスティックを使用
再プレイ……RUN (F・5)



待てい! チャンスのががらずにズドン!

爽快シューティングゲーム

```

100 '*** tank-2 ***
110 '
120 COLOR 15.4:SCREEN 1,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DIM S(2,13)
140 DEFFNR(A)=RND(1)*A:DEFFNP(X,Y)=VPEEK(6146+X+32*Y)
150 DEF FNX(S)=VAL(MID$("1122210001",S+1,1))-1
160 DEF FNY(S)=VAL(MID$("100122210",S+1,1))-1
170 FOR J=1 TO 5:S$="":FOR I=1 TO 8:READ A$:S$=S$+CHR$(V
AL("&H"+A$)):NEXT I:SPRITE$(J)=S$:NEXT J
180 FOR I=1 TO 2:FOR J=0 TO 13:READ S(I,J):NEXT J,I
190 '
200 PX(0)=1:PY(0)=2:PX(1)=26:PY(1)=21:PY(2)=209:PY(3)=20
9:FLG(0)=1:FLG(1)=1
210 LOCATE 0,1:PRINT STRING$(28,"□"):FOR I=2 TO 21:PRINT
"□":TAB(27);"□":NEXT:PRINT STRING$(28,"□")
220 FOR I=0 TO 50:LOCATE FNR(24)+2,FNR(18)+3:PRINT "●":N
EXT I
230 '
240 FOR PL=0 TO 1
250   S=STICK(PL):IF S<>0 THEN PA(PL)=S:FLG(PL)=0
260   X=PX(PL):Y=PY(PL):S=PA(PL):GOSUB 380:N=PL:C=10:GOS
UB 390:PX(PL)=X:PY(PL)=Y
270   IF FLG(PL)=0 AND STRIG(PL)=-1 AND PY(PL+2)>100 THE
N PA(PL+2)=PA(PL):PX(PL+2)=PX(PL):PY(PL+2)=PY(PL):SO=1:G
OSUB 370
280   X=PX(PL+2):Y=PY(PL+2):S=PA(PL+2):GOSUB 380:IF J=1
THEN PY(PL+2)=209:PUT SPRITE 2+PL,(0,209),1.5:GOTO 300
290   N=PL+2:S=9:C=9:GOSUB 390:PX(PL+2)=X:PY(PL+2)=Y
300   IF PX(0)=PX(3) AND PY(0)=PY(3) THEN PV(1)=PV(1)+1:
GOTO 330
310   IF PX(1)=PX(2) AND PY(1)=PY(2) THEN PV(0)=PV(0)+1:
GOTO 330
320 NEXT PL:GOTO 240
330 SO=2:GOSUB 370:PUT SPRITE 0,(8,208),1,1:CLS:GOSUB 36
0
340 IF PV(0)>4 OR PV(1)>4 THEN FOR I=0 TO 2000:NEXT:SOUN
D 7,248:PLAY "L6404BAGFEDC":END ELSE 200
350 '
360 LOCATE 1,0:PRINT"player-1 ";PV(0):TAB(15);"player-2
";PV(1):RETURN
370 FOR L=0 TO 13:SOUND L,S(SO,L):NEXT:RETURN
380 J=1:A=FNX(S):B=FNY(S):IF ABS(X*2-27)<26 AND ABS(Y*2-
23)<20 AND FNP(X+A,Y+B)=32 THEN J=0:X=X+A:Y=Y+B:RETURN E
LSE RETURN
390 PUT SPRITE N,(X*8+16,Y*8),C,(S+1)¥2:RETURN
400 DATA 18,18,18,FF,DB,5A,C3,FF,D8,F8,88,BF,BF,88,F8,D8
410 DATA FF,C3,5A,DB,FF,18,18,18,1B,1F,11,FD,FD,11,1F,1B
420 DATA 00,00,3C,3C,3C,3C,00,00
430 DATA 0,0,0,0,0,0,1,7,16,16,16,100,20,0
440 DATA 0,0,0,0,0,0,21,247,16,0,0,100,60,0

```


空軍指令「TOKIOヲ爆撃セヨ！」

8

BOMB-X

某国の誇る空軍所属B-52の爆撃機操縦士であるキミのもとに本国から1つの指令が届いた。「TOKIOヲ爆撃セヨ!」。これを受けたキミは即座に目的のTOKIOへ。しかし、あいにく空には雲が……。さて、キミは指令を遂行できるか?



PLAY




ブーンという音とともに、B-52が飛んでくる。中ほどに雲がフワフワと漂っている画面下に……ヤヤ、都市らしきものを発見! さて、そろそろ指令を遂行しようか……。おもむろにスペースバーを押す。爆弾は音をたてて落下したが、イカン! 雲が近づいてきたあーっ。

なんとか雲にあたらなかったものの、都市にもあたらずじまい。でも、まだ9発もあるもんネー。カーソルキーの左右で飛行機のを調整して、2発目発射! 今度は見事命中だ! ウウ、なんだかバク、この快感がクセになりそう……。

PROGRAM

アクションゲームでは非常に多く用いられるスプライト。いままでに何気なく使っていたスプライト割り込みを、少し説明してみよう。

スプライト割り込みとは、複数のスプライトパターンがぶつかったとき、プログラムの流れをその時点で強制的にあるルーチンに飛ばすもの。このプログラムでは280行がそのルーチンに当たる。このルーチン内で爆弾が雲とあたったか、都市とあたったかの判定をして、都市なら爆発音の発生、命中数UPをする。あとは、290行でいずれの場合も爆弾を画面上から消して180行へもどる。スプライトの割り込みを使ったからこそ、これだけ短くできたわけだ。



爆弾は放物線を描いて落ちる

爽快シューティングゲーム

```

100 ' *** bombx ***
110 '
120 COLOR 15,4:SCREEN 1,2:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DIM D(2,13):GOSUB 350
140 DEFBNX(S)=VAL(MID$("111211101",S+1,1))-1
150 B=10:X=3:YP=10:SO=1:GOSUB 330
160 ON SPRITE GOSUB 280:ON STRIG GOSUB 310:STRIG(0) ON
170 '
180 PUT SPRITE 2,(100+RND(1)*130,170),8.2:PUT SPRITE 0.(
XP,YP),14.0
190 '
200 S=STICK(0):DX=2*BNX(S):XP=(DX+XP+X) MOD 250:PUT SPRI
TE 0,(XP,YP),14.0
210 IF BF<>1 THEN 250
220 SO=2:GOSUB 330:SOUND 4,YB:YB=INT(YB*1.1):XB=(XB+BX)
MOD 250
230 IF YB>180 OR XB>255 OR CF=1 THEN CF=0:BF=0:SO=1:GOSU
B 330:IF B=0 THEN PLAY "V0":END ELSE STRIG(0) ON:PUT SPR
ITE 1,(0,209),0,1:GOTO 250
240 PUT SPRITE 1,(XB,YB),11,1
250 PUT SPRITE 3,(CX,120),15,3:CX=(CX+252) MOD 250
260 LOCATE 5,0:PRINT USING"BOMB = ## HIT = ##";B:HT:GO
TO 200
270 '
280 SPRITE OFF:IF YB<150 THEN CF=1 ELSE SO=0:GOSUB 330:F
OR W=0 TO 1000:NEXT:HT=HT+1
290 PUT SPRITE 1,(0,209),0,1:RETURN 180
300 '
310 STRIG(0) OFF:SPRITE ON:XB=XP:YB=36:BX=X+DX:BF=1:B=B-
1:X=X-B:RETURN
320 '
330 FOR S=0 TO 13:SOUND S,D(SO,S):NEXT:RETURN
340 '
350 FOR I=0 TO 3:S$="":FOR J=1 TO 32:READ A$:S$=S$+CHR$(
VAL("&H"+A$)):NEXT:SPRITE$(I)=S$:NEXT:FOR I=0 TO 2:FOR J
=0 TO 13:READ D(I,J):NEXT J,I:RETURN
360 '
370 DATA 60,50,66,50,60,00,00,CF,E8,E4,FE,FF,FF,00,00,00
380 DATA EE,82,EE,28,EE,00,00,00,80,4E,3F,1F,FE,00,00,00
390 DATA 00,04,06,06,07,03,01,01,03,07,06,06,06,07,03,01
400 DATA 00,90,B0,B0,70,60,40,C0,E0,F0,F0,F0,F0,E0,C0
410 DATA 00,00,00,00,1C,14,1C,15,1D,15,DD,D5,DD,FF,FF,00
420 DATA 08,08,08,1C,1C,14,1C,D4,5D,D5,5D,D5,5D,FF,FF,00
430 DATA 00,03,01,00,1C,3E,7F,E3,DF,FF,7F,3C,00,0E,3F,1C
440 DATA E6,FF,D8,00,00,38,FC,CE,BF,FF,FE,7C,00,00,00,00
450 '
460 DATA 0,0,0,0,0,0,20,247,16,0,0,100,60,0
470 DATA 30,0,30,0,30,0,31,240,28,0,0,80,0,8
480 DATA 0,0,0,0,0,0,0,56,14,14,14,200,15,11

```


9

効果音に凝るとこんなにエキサイティング

星間大戦争

詳細は省くが、とにかく旧地球軍と新スペースコロニー軍との全面戦争中である。地球軍は、コロニー軍の重要拠点を破壊しようとして、つぎつぎとミサイルを送りこんでくる。われわれは迎撃ミサイルで応戦。だが、ねらうのがやたらムズカシーんだ。



PLAY

左から右に飛んでくる敵のミサイルを、下から迎え撃つ。右端にわれらがコロニーがあるが、敵のミサイルを撃ちそこなうと、コロニーが1発で壊滅させられちゃう。




こいつは最初からむずかしいゲームだぜい。こちらの迎撃ミサイルの飛んでいく速さを把握していないと、敵のミサイルがのろいだけに、てきめんねらいをはずす。カーソルキーで迎撃ミサイルの発射位置の左右移動ができて、これを使用するのは初めのうちだけ。後になると敵のミサイルが速度を増して、移動している暇なんかアルワキャナイ！

PROGRAM

はっきりいって、このゲームは効果音がなければそれほどエキサイティングともいえない。ゲーム作りで効果音をうまく使うということは、そんなわけで重要な問題なのです。

サウンドを何回も利用する場合は、サウンドデータを配列に読みこんで置いて、サウンド出力サブルーチン呼び出すようにするのが、最もよい方法だ(450行)。

ちなみに基本的なことだけと忘れてならないのが、サブルーチン中で使用する変数はサブルーチン呼び出した時点で変わってしまうということ、注意してネ。



発射位置の調整がたいO K。よし発射！

爽快シューティングゲーム

```

100 ' *** gala-w ***
110 '
120 COLOR 15,1,1:SCREEN 2,2,0:KEY OFF:OPEN "GRP:" AS #1
130 DEFINT A-Z:R=RND(-TIME):DIM S(3,13)
140 FOR J=1 TO 2:S$=""
150   FOR I=1 TO 32
160     READ A$:S$=S$+CHR$(VAL("&H"+A$))
170     NEXT I
180     SPRITE$(J)=S$
190   NEXT J
200 SH=3:FOR J=1 TO 3:FOR I=0 TO 13:READ S(J,I):NEXT I,J
210 CO=7:GOSUB 460:GOSUB 370
220 '
230 A=STICK(0):IF J=0 THEN IF A=3 AND DX<240 THEN DX=DX+
8 ELSE IF A=7 AND DX>9 THEN DX=DX-8
240 IF STRIG(0) THEN S=2:GOSUB 450:J=1
250 IF DY<8 THEN DY=209 ELSE IF J=1 AND ABS(DY-80)<80 TH
EN DY=DY-8
260 PUT SPRITE 2,(DX,DY),2.2
270 GOTO 230
280 '
290 AX=AX+8:PUT SPRITE 1,(AX,AY),6,1
300 IF AX<200 THEN RETURN
310 S=1:GOSUB 450:SH=SH-1
320 IF SH>0 THEN CO=9:GOSUB 460:GOTO 210
330 SCREEN 1:PRINT "GAME OVER":PRINT "LEVEL ":LE
340 END
350 '
360 S=3:GOSUB 450:SPRITE OFF:INTERVAL OFF:PUT SPRITE 1,(
0,208),,1
370 AX=0:AY=RND(1)*88+8:DX=200:DX=160:DY=152:J=0
380 LE=LE+1:LINE(20,0)-STEP(200,7),1,BF
390 PRESET(20,0):PRINT #1.USING "LEVEL ##":LE;
400 PRINT #1.USING "   LEFT #":SH
410 ON SPRITE GOSUB 360:ON INTERVAL=40/LE GOSUB 290
420 INTERVAL ON:SPRITE ON
430 RETURN
440 '
450 FOR L=0 TO 13:SOUND L,S(S,L):NEXT:RETURN
460 CIRCLE(320,64),100,CO:PAINT (255,64),7:FOR W=0 TO 10
00:NEXT W:RETURN
470 '
480 DATA 00.00,00.00,00.00,00.00,E0.7F,C0.40,C0.40,C0.7F,E0.00
490 DATA 00.00,00.00,00.00,00.00,00.00,00.00,F8.06,01.06,F8.00,00.00
500 DATA 00.00,00.00,01.01,01.01,01.01,01.01,01.01,01.01,01.01
510 DATA 20.70,F8.F8,FC.FC,FC.FC,FC.FC,FC.FC,FC.FC,FC.FC,FC.FC,54.54
520 '
530 DATA 0.0,0.0,0.0,0.21,247.16,0.0,100.60,0
540 DATA 0.0,0.0,0.0,0.1,7.16,16.16,100.100,0
550 DATA 0.45,0.45,0.45,10.240,16.16,16.200,20,0


```


10

敵の動きを予測できるか？

追撃、戦車戦！

第二次世界大戦，フランス北東部。猛威をふるったドイツ軍も，わが合衆国の参戦で敗走を余儀なくされていた。が，その反撃力をあなだることはできない。いま，第24機甲師団に所属するワタシの戦車の前にも，鉄十字をつけたドイツ軍戦車が…。



PLAY

敵の動きはランダムで予想がつかない。これはアセル。なにしろ「TIME」が0になる前に敵を破壊しないとゲームオーバーなのだが，攻撃計画がたてられないのだ。だからといって，あわてて近づくと，それこそ危険……。



が，ランダムな敵の動きや攻撃にも，ある種の制限があることに気づけば，こっちのものだ。弾の速さから近づいても安全な範囲を考えたら，こちらは8方向に動けるが，敵は……，といったことを分析していけば勝利は目前。まあ，とにかくやってみるのがイチバンだけどネ。

PROGRAM

ここでは，配列 PX 、 PY に位置を，そして配列 A に向きを収納している。配列 PX 、 PY と配列 A の添字は共通しており，0 のときプレイヤーの戦車，1 のときそのミサイルを，2 のときコンピュータ側の戦車，3 のときそのミサイルを示す。

コンピュータ側の戦車をランダムに動かすには，カーソルキーで動かすかわりに，乱数発生させた数値を配列 A に入れてやる (380行)。

ちなみに，このプログラムでは，240行までがメインルーチンに入るまでの設定，250行から340行までがメインルーチン，そして350行からあとはサブルーチン群となっている。



人間味ってものがない敵の動きのランダムさよ

爽快シューティングゲーム

```

100 '*** tank-1 ***
110 '
120 COLOR 15,1,7:SCREEN 1,,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEFFNR(A)=RND(1)*A:DEFFNP(X,Y)=VPEEK(6146+X+Y*32)
150 DEFFNX(S)=VAL(MID$("112221000",S+1,1))-1:DEFFNY(S)=V
AL(MID$("100122210",S+1,1))-1
160 CLS:LOCATE 0,1:PRINT STRING$(28,"□")
170 FOR I=2 TO 21
180   PRINT "□";TAB(27);"□"
190 NEXT
200 PRINT STRING$(28,"□")
210 FOR I=0 TO 50
220   LOCATE FNR(24)+2,FNR(18)+3:PRINT "◆"
230 NEXT
240 LE=LE+1:TT=230-SQR(LE)*30:PX(0)=26:PY(0)=21:PX(2)=1:
PY(2)=2:A(1)=0:A(3)=0
250 FOR I=0 TO 3
260   X=PX(I):Y=PY(I)
270   ON I+1 GOSUB 350,360,380,390:X=X+FNX(A(I)):Y=Y+FNY
(A(I))
280   P=FNP(X,Y)
290   IF I MOD 2=1 AND A(I)>0 THEN IF P>16 AND P<21 THEN
IF I=1 THEN PLAY "L6404CDEFGAB":PO=PO+300:GOTO 160 ELSE
LOCATE PX(0),PY(0):PRINT "X":PLAY "L104C":END
300   IF P<>32 THEN X=PX(I):Y=PY(I):IF I MOD 2=1 AND A(I
)>0 THEN LOCATE X,Y:PRINT " ":A(I)=0
310   IF A(I)>0 THEN GOSUB 420
320 NEXT I
330 LOCATE 0,0:PRINT USING "TIME ###   SCORE #####":TT
,PO:TT=TT-1:PO=PO-1
340 IF TT<0 THEN FOR I=0 TO 10:PLAY "L64N20":NEXT I:END
ELSE 250
350 S=STICK(0):IF S=0 THEN RETURN ELSE A(0)=S:RETURN
360 IF STRIG(0)=0 OR A(J)>0 THEN RETURN ELSE PLAY "L64N8
0":A(I)=A(I-1)
370 X=PX(I-1):Y=PY(I-1):PX(I)=X:PY(I)=Y:RETURN
380 A(I)=FNR(4)*2+1:RETURN
390 IF (PX(2)<>PX(0) AND PY(2)<>PY(0)) OR A(I)>0 THEN RE
TURN ELSE PLAY "L64N80"
400 IF PX(2)=PX(0) THEN A(I)=SGN(PY(0)-PY(2))*2+3 ELSE A
(I)=SGN(PX(2)-PX(0))*2+5
410 X=PX(I-1):Y=PY(I-1):PX(I)=X:PY(I)=Y:RETURN
420 IF I MOD 2=1 THEN C=22+(A(I)¥2)MOD 2:GOTO 450
430 A=(A(I)¥2)MOD 4
440 IF A=0 THEN C=17 ELSE IF A=1 THEN C=20 ELSE IF A=2 T
HEN C=18 ELSE C=19
450 LOCATE PX(I),PY(I):PRINT " ":PX(I)=X:PY(I)=Y
460 LOCATE X,Y:PRINT CHR$(1)+CHR$(C+64)
470 RETURN

```


攻撃パターンをつかめ!

11 ベルリンの壁

いまや完全にコンピュータ化されている社会主義帝国と、われわれ自由主義連合は交戦中だ。キミの任務はベルリンの壁を守り、敵の侵入を防ぐこと。敵は壁を壊さずに攻撃できるエネルギー砲をもっている。この壁を守りきることができるか?



PLAY




敵とプレイヤーの間には、防御壁が2重に設置されている。プレイヤーは、この防御壁を壊してはじめて敵を攻撃できるわけだが、敵のエネルギー砲は、防御壁を通り抜けてくるのだ。不公平じゃないか、民主主義に反するゾ、と怒ってみてもはじまらない。

さて、ゲームスタート。スペースバーで壁を破壊し、敵めがけて発射だ。どうも敵さんは、おのれのエネルギー砲の力を過信しているのか、逃げるといふことをしないようだ。破壊された壁のすき間から、ソレ、発射。命中! ヌワンダ、軽いモンじゃないか。この敵の習性を利用すれば、かなり楽しく遊べるゾ。敗れてばかりのイライラゲーム、多いモンネ。

PROGRAM

使用変数について少し説明しておこう。プレイヤーのX座標はPXに、コンピュータ側はCXに入っている(共に横の移動のみなので、Y座標は定まっている)。こちらの発射したミサイルの座標は、M0、M1に、そしてコンピュータ側のミサイルの座標は、M2、M3に入っている。

コンピュータの動きは、280行で決めている。プレイヤーに近づいてくるか、ランダムに動くかは、280行のIF文で決めている。よりランダムにコンピュータ側を動かしたいときは、FNR(2)の数値2を大きくすればよい。



破壊した防御壁のすき間から確実に敵をねらえ

爽快シューティングゲーム

```

100 '*** kabe ***
110 '
120 COLOR 15.1.7:SCREEN 1,,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEFFNR(A)=RND(1)*A:DEFFNX(S)=VAL(MID$("111211101",S+
1.1))-1:DEFFNP(X,Y)=VPEEK(6146+X+Y*32)
150 '
160 CLS:SE=SE+1:CX=20:PX=5:M1=0:M3=0:TIME=0:TT=3000/(SE*
.3+2):ON INTERVAL=TT GOSUB 420
170 FOR I=6 TO 10 STEP 4
180   LOCATE 0,I:PRINT STRING$(28,"□")
190 NEXT:INTERVAL ON
200 '
210 PO=PO-1:LOCATE 0,0:PRINT USING "TIME #####":TT-TIME:
LOCATE 15,0:PRINT USING "SCORE #####":PO
220 LOCATE PX,22:PRINT " ":S=STICK(0):PX=PX+FNX(S)
230 IF ABS(PX*2-27)>27 THEN PX=PX-FNX(S)
240 LOCATE PX,22:PRINT "↑"
250 X=M0:Y=M1:J=STRIG(0):A=-4:M=PX:N=22:A$="I":GOSUB 35
0:M0=X:M1=Y
260 IF M0=CX AND M1=2 THEN INTERVAL OFF:FOR I=0 TO 6:PLA
Y "L3204B06B":NEXT:PO=PO+1000:FOR I=0 TO 3000:NEXT:GOTO
160
270 LOCATE CX,2:PRINT " "
280 IF FNR(2)>0 THEN CX=CX+FNR(3)-1 ELSE CX=CX+SGN(PX-CX
)
290 IF ABS(CX*2-27)>27 THEN CX=CX-SGN(CX)
300 LOCATE CX,2:PRINT "↑":J=0
310 IF PX=CX THEN J=-1
320 X=M2:Y=M3:A=5:M=CX:N=2:A$="I":GOSUB 350:M2=X:M3=Y
330 IF M2=PX AND M3=22 THEN INTERVAL OFF:PLAY "L6405B06B
07B04C":LOCATE 5,5:PRINT "GAME OVER":END ELSE 210
340 '
350 IF J=-1 AND Y=0 THEN X=M:Y=N
360 IF Y>0 THEN LOCATE X,Y:PRINT " " ELSE RETURN
370 Y=Y+A:IF ABS(Y-12)>10 THEN Y=0:RETURN
380 P=FNP(X,Y):LOCATE X,Y
390 IF P=219 THEN PRINT " ":BEEP:PO=PO+10:Y=0 ELSE PRINT
A$
400 RETURN
410 '
420 PLAY "04L64GFEDC"
430 LOCATE 5,5:PRINT "TIME OVER":LOCATE 5,0:PRINT "   0
"
440 END

```


12

これは貴族のスポーツ、盛装でプレイしろ！ ハンティングゲーム

きょうはハンティング。夜明け前から銃をたずさえて森へ繰り出す。来たかいがあった。沼の水面は眠って浮かんでいる鳥たちでいっぱい。やがて夜が明け、光が水面を照らすと、鳥たちは目覚め、一斉に飛びたつた。この瞬間を待っていたんだ。



PLAY

画面の右端に鳥が現れ、左に飛んでいく。カーソルキーで+印の照準を動かしてねらいを定める。よしまだ、スペースキー！ おっと素早くかわされた。よし、もう一度、発射。命中だ！ 撃ちのがした鳥が1羽でも画面の左端へ飛び去ったら、ゲームオーバー。ぐずぐずしてられない。かと思うと、1発発射するごとに1点ずつマイナスされるし、3発つづけてははずすとその場でゲームオーバー。慎重さも要するわけだ。

1面クリアして2面へ。こんどは鳥が増えた。全部撃ち落とせるだろうか。しまった、もう左の方へスイスイ飛んでいく鳥がいる。あわてて照準に追いかけてさせるが追いつかないじゃないか！




PROGRAM

改良法を紹介しよう。まず照準の移動量をもっと大きくしたい場合は、290行の10という値をもっと大きな数値に変えればよい。

スプライトデータを作ることができるようになったら、鳥をイノシシに変えるなんてこともできる。

また、照準が上端にある場合にさらに上向きのカールキーを押すと下端に、下端にある場合に下向きを押すと上端に照準が現れる、というふうにしたときは、290行のIF～文を $Y = (Y + 102)$

MOD132+30と変えるとよい。



ホラホラ、鳥が左端に達したらオシマイだぞ

爽快シューティングゲーム

```

100 '*** hunt-2 ***
110 '
120 COLOR 15,1:SCREEN 2,2:KEY OFF
130 DEFINT A-Y:R=RND(-TIME):DIM X(34),Y(34)
140 DEF FNR(A)=RND(1)*A
150 DEF FNX(S)=VAL(MID$("112221000",S+1,1))-1
160 DEF FNY(S)=VAL(MID$("100122210",S+1,1))-1
170 OPEN "GRP:" AS #1
180 FOR I=1 TO 2:S$=""
190   FOR J=1 TO 32
200     READ A$:S$=S$+CHR$(VAL("&H"+A$))
210   NEXT J:SPRITE$(I)=S$
220 NEXT I:NU=1
230 '
240 BH=0:GH=3:NU=NU+SGN(FIX(32/NU)):X=128:Y=86
250 FOR I=0 TO NU
260   X(I)=224:Y(I)=FNR(145)+16
270 NEXT I:LINE(14,30)-(242,178),2,B
280 '
290 S=STICK(0):X=X+FNX(S)*10:Y=Y+FNY(S)*10:IF ABS(Y-96)>
70 THEN Y=Y-FNY(S)*10
300 IF X<15 THEN X=15
310 IF X>225 THEN X=225
320 PUT SPRITE 0,(X,Y),8,2
330 IF STRIG(0)=-1 THEN PLAY "L8S11M300004C":GH=GH-1:PO=
PO-1:HJ=1:LINE(16,175)-(X+8,Y+8),8:LINE(239,175)-(X+8,Y+
8),8 ELSE HJ=0
340 IF GH<0 THEN 450
350 FOR I=0 TO NU
360   IF Y(I)=209 THEN 420
370   X(I)=X(I)-6*FNR(2):R=FNR(3)-1:Y(I)=Y(I)+10*R
380   IF X(I)<16 THEN 450
390   IF ABS(Y(I)-96)>66 THEN Y(I)=Y(I)-10*R
400   IF HJ=1 AND ABS(X-X(I))<16 AND ABS(Y(I)-Y)<16 THEN
PLAY "L804S14M200A":Y(I)=209:PO=PO+11:GH=3:BH=BH+1:HJ=0
410   PUT SPRITE I+1,(X(I),Y(I)),FNR(14)+2,1
420 NEXT I
430 LINE(16,175)-(X+8,Y+8),1:LINE(239,175)-(X+8,Y+8),1:L
INE(0,0)-(240,7),1,BF:PRESET (0,0):PRINT #1,"SCORE":PO;"
LEFT":GH:IF BH=NU+1 THEN PLAY "L6405CDEFG":GOTO 240 EL
SE 290
440 '
450 PLAY"L3203BAGFEDC":SCREEN 1:PRINT "GAME OVER":PRINT
"YOUR SCORE":PO
460 END
470 '
480 DATA 00,00,00,00,00,00,00,01,01,01,32,FC,FC,02,01,00
490 DATA 1F,20,43,80,83,84,8A,14,18,17,00,01,00,01,FF,00
500 DATA 01,01,01,01,01,01,01,FF,FF,01,01,01,01,01,01,01
510 DATA 80,80,80,80,80,80,80,FF,FF,80,80,p0,80,80,80,80

```


13

キミの反射神経を鍛えるジャンプゲーム

JUMP-JUMP

このごろ運動してないなあ、なんてキミにゼツタイおすすめするのが、この **JUMP-JUMP**。押しピンを避けて、ただひたすらジャンプするこのゲームで汗を流してほしい。プログラムも、プレイヤーの動きが連続でないために工夫がしてある。





PLAY

キミは反復横飛びを知っているかね？ そう、体育の時間でおなじみの運動能力テストの1つ。このゲームは、あの反復横飛び運動のゲーム版なのだ。

4本の道には、いたるところ押しピンが落ちている。キミはこのピンを踏まないように、右に左にジャンプするわけだ。

ジャンプはスペースバーで、方向はカーソルキーの左右で行う。このとき、カーソルキーをあらかじめ押しおき、ジャンプのタイミングを図って、バーをたたくのがコツ。ピンの数が少ないところをねらって……ジャンプ！


どれだけ長くピンを踏まずに飛べるか、運動神経を試すつもりでキミも挑戦してみよう。



PROGRAM

毎度おなじみスクロールゲーム。そう、プログラムの構成、手法は他のスクロールとまったく変わらない。ただ、プレイヤーのいる位置が4か所なので、プレイヤーの位置を表す変数 **PX** が画面上の位置 (X座標) ではなく、道の番号を表すなど少々工夫をしてある。

画面上の座標は、この変数をもとに計算している。270行と420行のスプライトのX座標、290行(実際には140行の **VPEEK** 文中)の関数 **FNP** も、これにもとづいている。



なるべく先までコースを読んでおくのがコツ

スクロールゲーム

```

100 ' *** Jump ***
110 '
120 COLOR 15,4,7:SCREEN 1,2,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEF FNP(X)=VPEEK(6309+X*7)
150 DEF FNX(S)=VAL(MID$("111211101",S,1))-1
160 '
170 S$=""
180 FOR I=1 TO 32
190   READ A$:S$=S$+CHR$(VAL("&H"+A$))
200 NEXT I
210 SPRITE$(0)=S$
220 '
230 PX=0:PY=5
240 PLAY "S4M300L6405"
250 ON STRIG GOSUB 400:STRIG(0) ON
260 '
270 PUT SPRITE 0,(PX*56+36,PY*8),15.0
280 GOSUB 350
290 IF FNP(PX)<>32 THEN PLAY "V1202GFEDBC":LOCATE 0,24:P
RINT "Your Score ";P:END
300 P=P+1
310 LOCATE 0,0:PRINT SPC(15);"Score ";P:SPC(5)
320 LOCATE 0,1:PRINT SPC(57)
330 GOTO 270
340 '
350 LOCATE 0,23:PRINT "~I   I~~I   I~~I   I~~I   I~";
360 LOCATE 3+INT(RND(1)*4)*7,23:PRINT "┘"
370 LOCATE 0,23:PRINT "~I   I~~I   I~~I   I~~I   I~"
380 RETURN
390 '
400 PLAY "DA"
410 S=STICK(0)+1:DX=FNX(S):PX=PX+DX:IF PX<0 OR PX>3 THEN
PX=PX-DX
420 PUT SPRITE 0,(PX*56+36,PY*8),15.0
430 RETURN
440 '
450 DATA 03,06,03,03,01,07,0B,13,53,03,06,0C,0C,06,03,0F
460 DATA E0,B0,E0,E0,C0,F8,E4,E2,E2,E0,39,18,18,30,60,78

```


14

風船にウラまれる覚えはないのに 風船の襲来

おやつ、向こうから飛んでくるのは……風船だ！ もちろん安心はできない。これはなんと爆弾なのだ。武器はレーザーガンだけ。急いで風船を撃ち落とさないとやられるぞ。割っても割っても風船はやってくる、しだいに速さを増しながら……。





PLAY

赤い風船が画面右端からやってきて左へ移動する。いままで紹介したゲームでは、敵は下から上に動いていたので、その点このゲームは異なったタイプともいえよう。また、ハンターはキーを押さなくても自動的に上へ動いていく。カーソル移動キーの押し方で、ハンターの動き方を調整するのだ。

ポイントは風船そのものをねらって撃つこと。上部をねらったほうがベターだ。風船のひもの部分を撃ってもほとんど効果はないので気をつけたまえ。

しだいに敵の動きが速くなると、ついあせてキー操作をまちがってしまうこともあるだろう。冷静かつ大胆にプレイしたまえ。




PROGRAM

まず風船の座標だが、これは配列 **BX**, **BY** に入っており、移動変化量は配列 **SX** に入っている。ハンターの座標は **UX**, **UY** に入っている。

カーソルキーの押し方によって、ハンターの動き具合が変わるので、一見むずかしいことをしているように見えるが、実はただ **STICK(0)** の値によって計算しているだけなのだ。360行の計算式がそれ。

<改良法> 風船の数は現在5個だが、この数を変えたいときは240行の **BN** の値を変えてやればよい。



ひもをねらっても風船は破裂しないぞ

スクロールゲーム

```

100 ' *** fusen-b2 ***
110 '
120 COLOR 15,1,1:SCREEN 2,2,0:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):OPEN "GRP:" AS #1
140 DEFFNR(A)=RND(1)*A
150 '
160 FOR J=1 TO 2
170   S$=""
180   FOR I=1 TO 32
190     READ A$:S$=S$+CHR$(VAL("&H"+A$))
200   NEXT I
210   SPRITE$(J)=S$
220 NEXT J
230 '
240 BN=4:FOR I=0 TO BN:BY(I)=209:SX(I)=4:NEXT
250 A=STRIG(0):IF A=-1 THEN LINE(180,UY+8)-(UX+16,UY+8),
15:PLAY "O4L64D"
260 FOR I=0 TO BN
270   IF A=-1 AND BX(I)<180 AND ABS(BY(I)-UY-4)<5 THEN P
O=PO+10:BY(I)=209:PLAY "O5L64B"
280   IF BY(I)<>209 THEN BX(I)=BX(I)-SX(I):GOTO 320
290   IF FNR(5)>0 THEN 330
300   BX(I)=240:BY(I)=16+FNR(160):SX(I)=SX(I)+2
310   IF SX(I)>16 THEN SX(I)=16
320   IF BX(I)<16 THEN 420
330   PUT SPRITE I+1,(BX(I),BY(I)),9,2
340 NEXT I
350 LINE(UX+16,UY+8)-(180,UY+8),1
360 UY=UY+STICK(0)*4-12:UY=(UY+192) MOD 192
370 PUT SPRITE 0,(UX,UY),2,1:LINE(132,0)-(255,7),1,BF
380 PRESET (132,0):PRINT #1,USING "SCORE #####";PO
390 LINE(UX+16,UY+8)-(180,UY+8),1
400 GOTO 250
410 '
420 SCREEN 1:WIDTH 29:PLAY "O4L16BAGDEFB"
430 LOCATE 0,5:PRINT "GAME OVER":PRINT "SCORE ";PO
440 END
450 '
460 DATA 3C.3C,3C,18,3C.3C,3C,3F,3F,3C,2C,66,66,62,C3,C1
470 DATA 00,00,00,00,00,FF,FF,E0,E0,C0,C0,00,00,00,00,80
480 DATA 3C,7E,FF,FF,FF,FF,7E,3F,03,01,01,00,00,00,00
490 DATA 00,00,00,00,00,00,00,00,00,00,80,C0,40,61,32,0C

```


15

ボクはあやしいヘリコプター調査団 鍾乳洞


鹿児島で世界一と思われる規模の大鍾乳洞が発見された。この洞くつは広大で、また深いと思われる。われわれ調査団は、勇敢にもこの未知の洞くつの中をヘリコプターで探検することにした。さあ出発！ キミはどこまで行けるかな？




PLAY

RUNしてまず最初に驚くのは、リアルなヘリコプターのサウンドだ。プル、プル、プルという飛行中の音、ぶつかったときの爆発音など、この効果音を聞いているだけでも思わず興奮してしまう。

鍾乳石が上から下から伸びていて、鍾乳洞の中は危険がいっぱいだ。ただカーソルキーの上下左右を押すだけでは、鍾乳石をかかわすことはできないゾ。そのためには少々工夫が必要だが……それは自分で見つけてみよう。ヘリコプターのプル、プル、プルを聞きながら、キミもあやしい調査団の一員になって、ゲームのテクを磨こうゼイ。



再プレイ……RUN (F・5)




PROGRAM

まずは、280行目と400行目の SOUND 文を見てもらいたい。これが、ヘリコプターと爆発音の正体だ。430/440行のデータから、音をつくりだしている。

もう1つのポイントは、380行目からの横のスクロールの部分。簡単にいってしまえば、文字列 PD\$ を単にプリントしているだけなのである。それぞれに配列 D\$() を割りあて、横にずらしては新しい部分をつけ加えて、これを合わせて PD\$ にし、それをプリントしている。

カーソルキーの操作は、ストリングを使った方法(関数 FNx, FNy)で行っていることにも注目。



うーむ快適な音だわい。そこへ巨大な鍾乳石が

スクロールゲーム

```

100 ' *** doukutu ***
110 '
120 CLEAR 800:SCREEN 1,0,0:COLOR 15,1,7:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEFFNR(A)=RND(1)*A
150 DEFFNP(X,Y)=VPEEK(6146+X+Y*32)
160 DEFFNX(S)=VAL(MID$("112221000",S+1,1))-1
170 DEFFNY(S)=VAL(MID$("100122210",S+1,1))-1
180 FOR I=1 TO 8
190   READ A$:S$=S$+CHR$(VAL("&H"+A$))
200 NEXT:SPRITE$(1)=S$
210 FOR I=0 TO 6:D$(I)=SPACE$(29):NEXT I
220 PX=0:PY=16:LE=LE+1:SC=SC+800
230 CLS:LOCATE 1,0:PRINT "LEVEL":LE:TAB(16):"Score":SC
240 ON INTERVAL=100/(1+LE*.2) GOSUB 320
250 FOR I=2 TO 21:LOCATE 0,I:PRINT STRING$(29,"□"):NEXT
260 FOR I=12 TO 18:LOCATE 0,I:PRINT SPACE$(29):NEXT
270 FOR I=0 TO 18:GOSUB 370:NEXT:INTERVAL ON
280 RESTORE 430:FOR I=0 TO 13:READ S:SOUND I,S:NEXT I
290 '
300 GOTO 300
310 '
320 S=STICK(0):PX=ABS(PX+FNX(S)):PY=PY+FNY(S)
330 PUT SPRITE 1,(PX*8+16,PY*8),6,1
340 IF PX>25 THEN INTERVAL OFF:PLAY "V0":PUT SPRITE 1,(0
,209),6,1:RETURN 210
350 GOSUB 400
360 SC=SC-1:LOCATE 21,0:PRINT SC
370 PD$="":CO=CO-1:IF CO<1 THEN CO=FNR(3)+4:LO=FNR(3)+3:
JU=FNR(2) ELSE IF CO<3 THEN JU=1:LO=10
380 FOR ID=0 TO 6:D$(ID)=RIGHT$(D$(ID),28)+"□":IF (JU=0
AND ID>LO) OR (JU=1 AND ID<LO) THEN MID$(D$(ID),29,1)="
"
390 PD$=PD$+D$(ID):NEXT ID:LOCATE 0,12:PRINT PD$:GOSUB 4
00:RETURN
400 IF FNP(PX,PY)<>32 THEN FOR I=0 TO 13:READ S:SOUND I,
S:NEXT:END ELSE RETURN
410 '
420 DATA 1F,04,04,8E,9F,FF,9F,0E
430 DATA 0,0,0,0,24,0,22,3,2,2,16,90,2,12
440 DATA 0,0,0,0,0,0,21,247,16,0,0,100,60,0

```


クギだらけの悪路をブットバセ!

16

ドライブゲーム

今日は待ちに待った休日だ。天気もいいし、隣に彼女を乗っけて、さあ、ドライブに出発だ。と、元気に出発したまではよかったが、近道をしたため、道のあちこちにクギが落ちている悪路にでってしまった。頼りはキミのドライブテクニックだけ!



PLAY




ど、どうして近道なんかしちゃったんでしょうね。RUNしてみるとわかるけど、なんともヒドイ道なんだよね、これが。あちこちにクギは落ちてるわ、グニャグニャと曲がりくねっているわで、いくらボクでも、すぐ事故ってしまうな〜。

なんて泣きごとをいってはいられない。さて、カーソルキーの左右で車を操作しながら、快適ドライブだ! おっと、急に道幅が狭くなったぞ、慎重に、慎重に……。ふーっ、なんとかきり抜けたと思ったら、今度は落ちていくクギの数がやたらと多い。左に移動、右に移動、また左に……。グエツ、クギを踏んでしまった! ……無理だよなー、初心者マークのボクにはこんな道、グッスン。

PROGRAM

このプログラムのポイントは、道路を作る方法にある。まず、230行目の変数 **TT** と **AD** に注目。AD は道の伸びる方向、TT はその方向へどのくらい伸びるかを定めるもの。つまり、道の曲がりと長さを決めているわけ。道幅の変化は、時間から決めているので周期的にせばまったり、広がったりする。

370行目で、スプライトのあるところに何かキャラクターがあったら、END へ。主なループは、230行から420行だ。



この悪路、キミのテクニックで突破できるかな

スクロールゲーム

```

100 ' *** drive ***
110 '
120 COLOR 2,11,4:SCREEN 1,2,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEFFNR(A)=RND(1)*A
150 DEFFNP(X,Y)=VPEEK(6146+X+Y*32)
160 DEFFNL$(X)=STRING$(X,"0")
170 FOR I=1 TO 32
180   READ A$:S$=S$+CHR$(VAL("&H"+A$))
190 NEXT I
200 SPRITE$(1)=S$
210 TIME=0:LE=3:WD=16:CX=10:LOCATE 0,10
220 '
230 TT=FNR(5):AD=FNR(3)-1
240 FOR I=0 TO TT
250   WX=WX+AD
260   IF WX<1 THEN WX=1:GOTO 280
270   IF WX+WD>26 THEN WX=WX-1:WD=26-WX
280   T=SGN(((TIME¥100) MOD 20)-10)
290   IF FNR(8)=0 AND ABS(WD+T-11)<7 THEN WD=WD+T
300   R$=FNL$(WX)+SPACE$(WD)+FNL$(28-WX-WD)
310   IF FNR(WD)>5 THEN MID$(R$,WX+FNR(WD)+1,1)="^"
320   LOCATE 0,23:PRINT R$
330   S=STICK(0)
340   IF S=3 AND CX<25 THEN CX=CX+1
350   IF S=7 AND CX>1 THEN CX=CX-1
360   PUT SPRITE 1,(CX*8+16.8),7.1
370   IF FNP(CX,3)+FNP(CX+1,3)=64 THEN 410
380   LE=LE-1:IF LE=0 THEN 440
390   LOCATE 0,0:PRINT USING "LEFT ##":LE:PLAY "L203C"
400   FOR W=0 TO 1000:NEXT W
410 NEXT I
420 GOTO 230
430 '
440 PLAY "O4L32BAGFEDC"
450 SCREEN 0:PRINT "YOUR SCORE ":TIME
460 END
470 '
480 DATA 05,E5,E7,FF,FF,E7,E6,06,07,07,67,7F,7F,67,07,03
490 DATA A0,A7,E7,FF,FF,E7,67,60,E0,E0,E6,FE,FE,E6,E0,C0

```


17

群がる風船爆弾をよけきれるか？

気球野郎、大冒険


5月16日 快晴 風向(北北西) 風速(5m/秒)

『冒険野郎, 気球で世界一周に挑戦!』と騒がれて出発し, はや数か月。が, なんとイ・イ戦争まっ最中のイラン上空にきちまった。スパイとまちがえて, 風船爆弾を飛ばしてきたぞー!



PLAY

左右のカーソル移動キーを自在に操り, 下から昇ってくる風船爆弾を右に左に避けていく。黒い風船をとればボーナス点なのだが, それどころじゃない。だんだん右に追いつめられてきた。うっ, このままではヤラレル, えーい, もっと右へ, とさらにカーソルキーを押すと, なんと反対側の左端から気球が出てきたではないか! ラッキー! と油断してたらやられてしまった, ドーシテ, ボクが何をしたってゆーの?





PROGRAM

短いプログラムだが, 220行までが初期設定。230行から320行までがメインプログラムだ。得点となる風船の出現率を高めたいときは, 270行の後ろのほうの IF 文の数値 6 を, 小さくしてやればよい。

画面の端に気球をもっていっていったとき, さらにその方向に動かし続けると, 反対側の端から出現する。これを可能にするのが, 250行の $KX = (KX + AX + 256) \text{MOD} 256$ である。これは, いろんなゲームに応用できるので便利だ。覚えておこう。

気球の移動をもっと速くしたい人は, 240行の AX の値(±4)の絶対値を大きく, 遅くしたい人は小さくしてやればよい。



黒い風船だ! 取ればボーナス点なんだけど…

スクロールゲーム

```

100 ' *** baloon ***
110 '
120 COLOR 15,1,0:SCREEN 1,2:WIDTH 32:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEFFNR(A)=RND(1)*A
150 DEFFNP(X)=VPEEK(6146+X+32)
160 DEFFNX(S)=VAL(MID$("444844404",S+1,1))-4
170 LE=5:KX=104:AX=4
180 FOR I=1 TO 32
190   READ A$:S$=S$+CHR$(VAL("&H"+A$))
200 NEXT I
210 SPRITE$(1)=S$
220 '
230 PO=PO+1:LOCATE 15,0:PRINT USING "Score #####":PO
240 S=STICK(0):IF S<>0 THEN AX=FNX(S)
250 KX=(KX+AX+272)MOD272:PUT SPRITE 1,(KX+16.8),7,1
260 IF PO MOD 50=0 THEN LE=(LE-1)*SGN(LE-1)
270 IF FNR(LE)>1 THEN P$=" " ELSE IF FNR(6)>0 THEN P$="
●1" ELSE P$="○1"
280 X=FNR(28)
290 LOCATE X,23:PRINT LEFT$(P$,1)
300 LOCATE X,23:PRINT RIGHT$(P$,1)
310 IF SGN(AX)<0 THEN X=KX¥8 ELSE X=KX¥8+1
320 V1=FNP(X):V2=FNP(X+1):IF V1=132 OR V2=132 THEN PLAY
"L6406CEA":PO=PO+60:GOTO 230 ELSE IF V1<>133 AND V2<>133
THEN 230
330 PLAY "04L64BAGFEDC"
340 LOCATE 0,5:PRINT USING "Score #####":PO
350 END
360 '
370 DATA 03,0F,3F,7F,FF,FF,FF,84,44,24,12,0A,05,07,07,07
380 DATA C0,F0,FC,FE,FF,FF,FF,21,22,24,48,50,A0,E0,E0,E0

```


18

白銀の世界で、壮快なシュプールをえがこう！ スキーゲームPART1

さーて、こんどはお待ちかね。＼白銀は招くよ＼パソコンも招くよ、ってわけでスキーゲームなのだ。プログラムをキーでたたいて打ちこむのは、けっこうメンドーなもの。だけど苦あれば楽あり。苦労して打ちこんで、さあ爽快にすべろうぜい。



PLAY

RUN! プログラムを打ちこんで、この **RUN** させるときが、やっぱりイチバンの感動だね。つい♪ランラン♪とハナ歌が出たりして……。

それはともかく、ゲームが始まった。画面上方にスキーヤー、そして下のほうには赤い旗や青い旗。




◀キーを押すと、スキーヤーが左に動かかわりに、旗のほうが右側に移動する。ナール、こうして動かしてスキーヤーをこの旗にぶつけていけば得点になるというわけか。ヨーシ、ほれ、あちよい……おっ、赤い旗を取ったら得点がグリーンと伸びた。ねらいは赤い旗だな、よし今度こそ、えっもう時間？ 1分しか時間がないとはキビシーなあ……。ではもう一度、ほれ、あちよい……。

PROGRAM

配列 **XF**, **YF** は旗の位置, **PF** は、旗のポイントのための配列だ。

スプライトパターンは全部で3つあるが、はじめの2つはスキーヤー用。左と右へ行く方向によって変えている。最後の3つ目は、旗のスプライトだ。

プログラムの主な部分は、290~390行目。旗の移動、表示、判断をぐるぐると続けていて、1分間たつと、自動的に **GAME OVER** となるようにするのが230行目の **ON INTERVAL**~の命令だ。



旗は、よけるのではなくぶつかれば得点なのだ

スクロールゲーム

```

100 ' *** ski-1 ***
110 '
120 COLOR 1.15.15:SCREEN 1.2.0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEF FNR(A)=INT(RND(1)*A)
150 DEF FNX(S)=VAL(MID$("111211101",S+1.1))-1
160 FOR J=1 TO 3
170   S$=""
180   FOR I=1 TO 32
190     READ A$:S$=S$+CHR$(VAL("&H"+A$))
200   NEXT I
210   SPRITE$(J)=S$
220 NEXT J
230 XF=1:ON INTERVAL=3600 GOSUB 410:INTERVAL ON
240 FOR I=0 TO 4
250   PF(I)=1
260 NEXT I
270 PF(5)=5
280 '
290 FOR I=1 TO 5
300   XF(I)=XF(I)+XF*8:YF(I)=YF(I)-14
310   IF ABS(XF(I)-127)<=128 AND ABS(YF(I)-95)<=94 THEN 3
320   IF FNR(10)>7 THEN XF(I)=FNR(241):YF(I)=192 ELSE YF(
I)=209
330   PUT SPRITE I+1,(XF(I),YF(I)),4+PF(I),3
340   IF ABS(XF(I)-105)<13 AND ABS(YF(I)-42)<13 THEN YF(
I)=209:BEEP:PO=PO+100*PF(I)
350 NEXT I
360 S=STICK(0):IF S<>0 THEN XF=FNX(S)
370 PUT SPRITE 1,(104,40),2,(3-XF)/2
380 LOCATE 0,0:PRINT USING "Score #####":PO
390 GOTO 290
400 '
410 PLAY "L6404GABBBEDCC"
420 LOCATE 5,8:PRINT "TIME OVER"
430 END
440 '
450 DATA 07.07.03.03.07.0F.3E.31.11.13.13.09.09.83.7F.30
460 DATA 00.00.00.00.E0.B0.D8.C8.88.04.05.8F.FE.E2.02.00
470 DATA 00.00.00.01.07.0D.1B.13.13.11.C8.F8.3D.0F.0B.08
480 DATA 70.70.60.E0.F0.F8.CC.84.84.C4.E8.E8.C8.89.E3.7E
490 DATA 00.00.01.07.3F.7F.03.00.00.00.00.00.00.00.00
500 DATA 1C.7C.FC.FC.FC.FC.FC.FC.3C.0C.0C.0C.0C.0C.0C

```


19

回転競技でキミも見栄スキーヤーだ! スキーゲームPART2


PART 1 が初心者用ゲームであったのに対して、PART 2 は中級者用ゲームだ。というのも、木にあたってはいけないし、旗も全部とらないといけないからだ。でも、使用するキーはたったの1つ。操作がヒジヨーに単純なのがウレシイネ。



PLAY

PART 1 と、どこがちがうのかなあ……と、さっそく RUN させてみることにした。おや、今度はスキーヤーが左右に動くぞ、しかも、カーソルキーを使うのではなくスペースバーを押すのだ。


スペースバーを押すと、いままでと逆方向に動きだすようになる。しかも、旗はすべて取らなくてはいけないのに、アチコチにある木が障害物になって取りにくいこと……。あつ、右端の旗を取ったあとは、左端の旗だ。ウワー、木にあたってゲームオーバー！ 最初にスピードを聞いてくるが、遅いからやさしいというわけではないようだ。



PROGRAM

スプライト表示は細かい16×16のモード。配列 X (2, 4), Y (2, 4) は旗の位置を覚えておくためのもの。変数 LE は速度を決めるものだが、同時に表示される障害物の数も決めていることに注目。つまり、表示の個数が少なくなると、速度も速くなるというわけだ。プログラムの中心部分は、280行から360行目の GOTO までのループ。障害物のスプライトにぶつくと380行目のサブルーチンに飛ぶが、ここではどのスプライトにぶつかったかをチェックして、それを消している。

ちょっとしたテクが変数 CC。これによって、障害物が重なってでるのを防いでいる(280行と300行)。



ウワツ、目の前に木！ よけられるかっ！

スクロールゲーム

```

100 ' *** ski-2 ***
110 '
120 COLOR 1,15,15:SCREEN 1,2,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DEFFNR(A)=RND(1)*A
140 FOR J=0 TO 3
150   S$=""
160   FOR I=1 TO 32
170     READ A$
180     S$=S$+CHR$(VAL("&H"+A$))
190   NEXT I
200   SPRITE$(J)=S$
210 NEXT J
220 LOCATE 0,5:INPUT "Speed 0(fast)-4(slow)":LE
230 IF LE<0 OR LE>4 THEN 220
240 AD=6:PY=40:CLS
250 FOR I=0 TO 1:FOR J=0 TO LE:Y(I,J)=209:NEXT J,I
260 ON SPRITE GOSUB 380:SPRITE ON
270 ON STRIG GOSUB 440:STRIG(0) ON
280 LOCATE 0,0:PRINT "SCORE ":PO:CC=CC-1
290 FOR J=0 TO 1:FOR I=0 TO LE
300   IF Y(J,I)=209 AND FNR(2)=0 AND CC<0 THEN CC=2:X(J,
I)=FNR(230):Y(J,I)=192
310   IF Y(J,I)<8 THEN IF J=1 THEN 450 ELSE Y(J,I)=209 E
LSE IF Y(J,I)<209 THEN Y(J,I)=Y(J,I)-8
320   PUT SPRITE 1+J*10+I,(X(J,I),Y(J,I)),3+5*J,J
330   IF PX<6 THEN AD=6 ELSE IF PX>230 THEN AD=-6
340   PX=PX+AD:PUT SPRITE 0,(PX,PY),5,AD/12+2.5
350 NEXT I,J
360 GOTO 280
370 '
380 STRIG(0) OFF
390 FOR J1=0 TO 1:FOR I1=0 TO LE
400   IF Y(J1,I1)=209 OR ABS(X(J1,I1)-PX)>18 OR ABS(Y(J1
,I1)-PY)>18 THEN 420
410   IF J1=0 THEN 450 ELSE PLAY "L6406A":PO=PO+100:Y(J1
,I1)=209:PUT SPRITE 1+J1*10+I1,(0,209),1,1:STRIG(0) ON:R
ETURN
420 NEXT I1,J1
430 STRIG(0) ON:RETURN
440 AD=-AD:RETURN
450 PLAY"L102A":END
460 '
470 DATA 25.F3.29.D5.93.79.67.91.0F.15.23.01.01.01.01.01
480 DATA 94.94.BF.C2.86.9B.E4.8C.92.A1.C0.80.80.80.80.80
490 DATA 40.40.7F.7F.7F.7E.70.40.40.40.40.40.40.40.40
500 DATA 00.00.FF.FC.C0.00.00.00.00.00.00.00.00.00.00
510 DATA 01.01.00.01.01.01.03.05.03.03.06.0C.06.02.0F.F8
520 DATA C0.C0.80.C0.F0.C8.C8.D0.C8.04.82.41.2F.F8.80.00
530 DATA 01.01.00.01.07.09.09.09.05.02.04.08.F1.3F.45.80
540 DATA C0.C0.80.C0.C0.E0.D0.C8.D0.E0.70.98.30.60.F0.1F

```


20

冗談から棒が出たケツサク・ゲーム やぶから棒

のどかな春の日、ワタシことウサギのウサオちゃんは、のんびりと竹やぶの中を散策していた。と、ヤブカラボウに、やぶから棒が出てきた。人間の子どもがイタズラしてるにきまっている。アツタマにきたけど、とにかく逃げなくっちゃ!



PLAY

たいていは、ゲームを作ってから題名を考えるのだけど、このゲームは先に題名が出てからゲームを考えたもの。なかなかスルドイでしょ(ほとんど冗談で作っている、という声もあるが)。

カーソルキーでウサギを動かして、ヤブからぐいっと突き出てくる棒をよけ、進んでいく。棒が不意に伸びたり縮んだりするので、先を読みながらすばやく進まないといけな。




それでも3面ぐらまでは、ウサギも機敏に動き、それほど苦労しないでも進めるはず。これが4面になると、さすがのウサオちゃんも疲れが見えて、動きが鈍くなるのだ。オっと危ない、こりゃー散歩も命がけだなや。

PROGRAM

棒がぐいぐい、によきによきと出てくる感じを出すために、BJ(IC)というカウンターを使っている。これで、描く線の長さを決めてあるわけである(370行)。

棒の伸び方がノロイ! と思った人は、330行の $BJ(IC) = FNR(3) + 3$ の値を減らせばよい。ただし1以上でね。

それにしても、このウサオちゃんのキャラクターなかなかよくできてるでしょ(と自画自賛するワタシであった)。



棒のツツ、とかスー、とかいう動きがやらしい

絶妙コントロールゲーム

```

100 ' *** yabu ***
110 '
120 WIDTH 29:COLOR 3,11,7:SCREEN 2,2,0:KEY OFF:OPEN "GRP
:" AS #1
130 DEFINT A-Z:R=RND(-TIME):DEFFNR(A)=RND(1)*A
140 FOR I=1 TO 32
150 READ A$:S$=S$+CHR$(VAL("&H"+A$))
160 NEXT I
170 SPRITE$(1)=S$:LE=0
180 FOR J=0 TO 100:PRESET(FNR(256),FNR(32)+8+FNR(2)*152)
:COLOR 2:DRAW "D8":NEXT:PLAY "O4L64CBDAEGF"
190 '
200 SC=SC+500:PX=40:PY=96:LE=LE+1
210 LINE(0,0)-(255,7),11,BF:PSET(0,0),1:PRINT #1.USING
"LEVEL ## ":LE::PRINT #1," SCORE ":SC
220 ON INTERVAL=(LE-10)*2*(LE<10)+1 GOSUB 300:INTERVAL O
N
230 '
240 INTERVAL STOP:S=STICK(0):IF (S=3) OR (S=7 AND PX>8)
THEN PLAY "L64N80":PX=PX+4*(5-S)
250 PUT SPRITE 1,(PX,PY),15,1
260 IF PX>240 THEN INTERVAL OFF:PLAY "O4L64CDEFGAB":FOR
I=0 TO 1000:NEXT:GOTO 200
270 SC=SC-1:LINE(160,0)-(240,7),11,BF:PSET(160,0),8:PRIN
T #1,SC
280 INTERVAL ON:GOTO 240
290 '
300 INTERVAL OFF
310 FOR IC=0 TO 1
320 IF BJ(IC)>0 THEN 350
330 BX(IC)=FNR(60)+PX-10:BJ(IC)=FNR(3)+3
340 IF FNR(2)=0 THEN BY(IC)=40:BL(IC)=80 ELSE BY(IC)=1
52:BL(IC)=-80
350 PLAY "L64N28":LINE(BX(IC),BY(IC))-(BX(IC)+1,BY(IC)
+BL(IC)/BJ(IC)),8,BF
360 IF ABS(PX+7-BX(IC))<8 AND (PY+7-BY(IC))*(PY+7-BY(I
C))-BL(IC)/BJ(IC)<0 THEN 420
370 IF BJ(IC)=1 THEN LINE(BX(IC),BY(IC))-(BX(IC)+1,BY(
IC)+BL(IC)),11,BF
380 BJ(IC)=BJ(IC)-1
390 NEXT IC
400 INTERVAL ON:RETURN
410 '
420 FOR I=1 TO 4:PLAY "O4L32GFEDC":NEXT:SCREEN 1:PRINT
"GAME OVER. SCORE ":SC:END
430 '
440 DATA 07,01,00,00,00,00,DF,E0,E0,40,40,40,40,4F,20,1F
450 DATA F0,F8,3C,0E,0E,12,E1,25,11,09,0E,18,08,24,F4,0C


```


21

銀河一のおそーじ名人たあオレのことヨ

ブラックホールの掃除人

全銀河系に支店を持つわが垂随喜社は「宇宙の便利屋」として有名だ。今回の仕事はちよいとやっかいで、第471ブラックホールのまわりの小惑星群をかたづけること。しかも、ライバル会社が付近に機雷を設置したという情報もはいつているのだ！



PLAY

制限時間の1分20秒以内に、できるだけ多くの星(・で表示)をブラックホール(□で表示)の中に入れるというゲーム。あちこちに機雷(◆で表示)がちらばっており、その中のひとつは自動追尾装置つきだ。触れたらイチコロである。




移動は上下左右にできるが、星を動かすには押す一手、ひっばるのは不可能。また1個ずつしか動かせない。また、押した星が機雷に触れてもゲームオーバーなので、要注意！

PROGRAM

<改良法> 自動追尾装置付きの機雷は、あっちへフラフラ、こっちへフラフラしながら近づいてくるのだが、さらにフラフラ度を増したければ、290行のFNR(9)の数値9をもっと大きくとればよい。また、表示される爆弾や星の数を、それぞれ変えるには、190行のFNRの中の数値を変えればよいのである。

また、M、Nが現在いる位置であり、X、Yは、進行方向上のいちばん手前の座標を、S、Tはそのひとつ先の座標を示している。

ちなみに、星の“・”はピリオドではなく、キャラクターコード表でいうと165なので注意すること。



ぐずぐずしていると機雷がフラフラ寄ってくるぞ

絶妙コントロールゲーム

```

100 '*** blckhll ***
110 '
120 COLOR 15,1,5:SCREEN 1,0:WIDTH 29:KEY OFF:DEFINT A-Z:
R=RND(-TIME):ON ERROR GOTO 380
130 DEFFNR(A)=RND(1)*A+1:DEFFNP(X,Y)=VPEEK(6146+X+Y*32)
140 DEFFNX(S)=VAL(MID$("112221000",S+1,1))-1:DEFFNY(S)=V
AL(MID$("100122210",S+1,1))-1
150 FOR I=1 TO 8
160   READ A$:S$=S$+CHR$(VAL("&H"+A$))
170 NEXT I
180 SPRITE$(1)=S$
190 A=FNR(35):A$="◆":GOSUB 330:A=FNR(5)+20:A$="·":GOSUB
330
200 LOCATE 16,12:PRINT "D":E=27:F=21:N=1:M=0
210 ON INTERVAL=4800 GOSUB 370:INTERVAL ON
220 S=STICK(0):A=FNX(S):B=FNY(S):X=M+A:Y=N+B:S=M+2*A:T=N
+2*B:P=FNP(X,Y):Q=FNP(S,T)
230 IF ABS(X-14)>14 OR ABS(Y-11)>10 THEN 260
240 IF P=165 THEN IF Q=32 THEN M=X:N=Y:LOCATE M,N:PRINT
" ":LOCATE S,T:PRINT "·" ELSE IF Q=219 THEN PLAY "L6404C
DEFG":PO=PO+50:M=X:N=Y:LOCATE X,Y:PRINT " " ELSE IF Q=13
1 THEN 380
250 IF P=131 THEN 380 ELSE IF P=32 THEN M=X:N=Y
260 PUT SPRITE 1,(M*8+16,N*8),3,1
270 LOCATE 8,0:PRINT USING "POINT #####":PO
280 PLAY "L64N30"
290 IF FNR(9)>2 THEN X=E+FNR(3)-2:Y=F+FNR(3)-2 ELSE X=E+
SGN(M-E):Y=F+SGN(N-F)
300 IF ABS(X-14)>14 OR ABS(Y-11)>10 OR FNP(X,Y)<>32 THEN
220
310 LOCATE E,F:PRINT " ":E=X:F=Y:LOCATE E,F:PRINT "◆"
320 IF E=M AND F=N THEN 380 ELSE 220
330 FOR I=0 TO A
340   LOCATE FNR(25)+1,FNR(19)+2:PRINT A$
350 NEXT I
360 RETURN
370 LOCATE 5,5:PRINT "TIME OVER":PLAY "L203B":END
380 COLOR 9:LOCATE 5,5:PRINT "GAME OVER"
390 FOR I=1 TO 5
400   PLAY "L6404BAG'EDC"
410 NEXT
420 END
430 DATA 38,38,10,38,7C,BA,28,28


```


22

警視庁も推薦の交通安全ゲーム!?

パッシング・ロード

ボクは男の友だちとさえない顔で、渋谷の公園通りを歩いていた。ふと、通りの反対側を見ると、あこがれの聖子ちゃんがボクらに手を振っているではないか……。早く道を渡らなければ、彼女を見失ってしまう。車を避けて逃げ!





PLAY

RUNすると、画面中央には道路が、左側にはキミが操作する人間たちが現れる。ウーム、なかなかキレイな道路だワイ、渡るのがもったいない(?) などといった暇はない。さあ、PLAY開始だ。

車は道路の右半分が上から下、左半分が下から上へ移動している。1面はなんなくクリア。2面以降は、車の数も増え、人の動きも心なしか鈍くなるようだ。

車の型はみな同じだが、1台1台の色がさまざまで、とってもカラフル。でも、車にばかり見とれていると、しだいに難易度を増すこのゲーム、すぐGAME OVERになっちゃうぞ!




PROGRAM

まず、使っている変数の説明から……。初めに宣言してある配列はCX, CYが車の位置、CCが車の色だ。AYはそれぞれの車の進む方向を示している。

変数を初期化、背景画面をつくったあと、プログラムは320行目の人を動かす部分を実行し、380行目からの車を動かす部分を実行する。人が車とぶつくと440行に飛んできてGAME OVER。

ちなみに、レベルを示すLEが増えていくと、NJが増えて、同時に現れる車が多くなってくる。



「バックヤロー!! ウロウロするな!」

絶妙コントロールゲーム

```

100 '*** p-road ***
110 '
120 COLOR 15,1:SCREEN 2,2:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DIM CX(31),CY(31),CC(31),AY(
31)
140 DEF FNR(A)=RND(1)*A
150 DEF FNX(S)=VAL(MID$("112221000",S+1,1))-1
160 DEF FNY(S)=VAL(MID$("100122210",S+1,1))-1
170 OPEN "GRP:" AS #1
180 FOR I=0 TO 2:S$=""
200   FOR J=1 TO 32
210     READ A$:S$=S$+CHR$(VAL("&h"+A$))
220   NEXT
230   SPRITE$(1)=S$
240 NEXT I
250 LINE(75,0)-(166,191),14,BF:LINE(119,0)-(122,191),10,
BF:FOR J=0 TO 1:FOR I=0 TO 5:LINE(97+44*J,32*I)-(100+44*
J,16+32*I),15,BF:NEXT I,J
260 FOR I=1 TO 31:CY(I)=209:NEXT
290 ON SPRITE GOSUB 440:SPRITE ON:PLAY "T255"
300 PLAY "L6404CDEFGAB":LE=LE+1:NJ=NJ+SGN(INT(31/LE)):PX
=40:PY=80
310 LINE(0,0)-(64,16),1,BF:PRESET(10,0):PRINT #1,"LEVEL"
:LE
320 PE=1-PE:S=STICK(0):PX=PX+FNX(S)*4:PY=PY+FNY(S)*4
330 IF PX<12THEN PX=12
340 IF PY<0 THEN PY=0
350 IF PY>174THEN PY=174
360 PUT SPRITE 0,(PX,PY),9,PE:PUT SPRITE 1,(PX-16,PY),8,
1-PE
370 IF PX>200 THEN 300
380 CT=CT-1
390 FOR IC=1 TO NJ
400   IF CY(IC)<>209 THEN IF ABS(95-CY(IC)-AY(IC))>95 TH
EN CY(IC)=209 ELSE CY(IC)=CY(IC)+AY(IC) ELSE IF CT<0 THE
N GOSUB 430
410   PUT SPRITE IC+1,(CX(IC),CY(IC)),CC(IC),2
420 NEXT IC:GOTO 320
430 A=FNR(4):CT=2:CC(IC)=FNR(13)+1:CX(IC)=80+22*A:AY(IC)
=8:CY(IC)=0:IF A<2 THEN AY(IC)=-8:CY(IC)=190:RETURN ELSE
RETURN
440 PLAY "T200L6404DRDRDR32DRFRER8ERDR16DR16C#.R8D":FOR
W=0 TO 2500:NEXT W:END
450 '
460 DATA 01,01,01,00,01,03,05,09,09,09,01,01,01,0F,0C,0E
470 DATA C0,C0,C0,80,C0,E0,D0,CE,C0,C0,C0,C0,E0,B0,30,3C
480 DATA 01,01,01,00,01,01,01,01,01,01,01,01,01,01,01,01
490 DATA C0,C0,C0,80,C0,C0,C0,C0,C0,C0,C0,40,40,40,E0
500 DATA 07,08,38,38,3F,08,08,09,09,08,08,3F,38,38,08,07
510 DATA E0,10,1C,1C,FC,10,10,90,90,10,10,FC,1C,1C,10,E0

```

23

台風襲来, 小島に注意し緊急入港せよ。

緊急入港ゲーム

台風15号が接近しつつあった。中心気圧 890 mb, 最大風速 80mの超大型台風だ。さしもの大型客船 MSX 号も緊急入港の必要にせまられた。しかし, もよりの港の付近は, いたるところに小島や岩があるのだ。はたして無事入港できるか?



PLAY




船は自動的に進んでいくので, カーソルキーで上下に動かし, 左右キーで加速, 減速して障害物をさける。島や陸地に少しでも触れるとゲームオーバーになるため, どのくらいまで近づいても大丈夫なのかという感覚をつかむのに少々時間がかかるようだ。

面が進むにつれ, 船の速度が速くなるので, 島をかわして入港するのが非常にむずかしくなる。その上, ばいきんのように動き回る機雷にふれてもいけない。3面まではどうにかクリアできるのだが, 4面でいつもしくじってしまうのだ。くやしーッ。

PROGRAM

スプライトは, 8×8 とか 16×16 とか正方形なので, たて長とかよこ長などのものを移動させたい場合には, スプライトを2つとか3つ並べて表示させる。このゲームでは, 8×8 のスプライトを3つ並べて船を表示している。

このため, 船の移動先に障害物があるかどうかの判断についても, 3つのスプライトの場所, つまり3か所のキャラクターを調べる。ただし, これは, 船のY座標がキャラクターの大きさの8で割りきれるときだけ。その他は, 3つのスプライトが6つのキャラクターにまたがるので, その6つのキャラクターを調べるようにしてある。これで, 少しでも触れようとすると, あたったと見なされることになるのだ。



島をよける! 動き回る機雷に気をつけろ!

絶妙コントロールゲーム

```

100 * *** ship ***
110 *
120 COLOR 15,4:SCREEN 1,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEFFNR(A)=RND(1)*A:DEFFNX(S)=VAL(MID$("112221000",S+
1,1))-1:DEFFNY(S)=VAL(MID$("100122210",S+1,1))-1:DEFFNP(
X,Y)=VPEEK(6146+X+Y*32)
150 FOR I=1 TO 4
160   S$=""
170   FOR J=1 TO 8
180     READ A$:S$=S$+CHR$(VAL("&H"+A$))
190   NEXT J
200   SPRITE$(I)=S$
210 NEXT I
220 ON SPRITE GOSUB 450:SPRITE ON
230 *
240 FOR I=0 TO 4:LOCATE FNR(15)+6,FNR(21)+1:PRINT "□":NE
XT I
250 SE=SE+1:LOCATE 0,0:PRINT "LEVEL";SE
260 FOR I=0 TO 21:LOCATE 25,I:PRINT STRING$(4,"□"):NEXT
I
270 FOR I=0 TO 1:A=FNR(8)+I*11+1:LOCATE 25,A:PRINT SPC(4
):LOCATE 25,A+1:PRINT SPC(4):NEXT I
280 SX=16:SY=88:GX=168:GY=88:PLAY"L4S14M20004A":FOR I=0
TO 2000:NEXT:AD=SE MOD 16+1
290 *
300 PLAY "L6404C":S=STICK(0):SX=SX+AD+FNX(S)*AD/2:IF SX>
240 THEN PUT SPRITE 0,(0,208),15,0:PLAY "L207F":FOR I=0
TO 3000:NEXT I:GOTO 240
310 SY=((SY-8)+FNY(S)*2+160) MOD 160)+8
320 FOR I=0 TO 2
330   PUT SPRITE I,(SX-I*8,SY),9,I+1
340   X=(SX-8*I)¥8-1:Y=SY¥8:GOSUB 440
350   IF SY MOD 8<>0 THEN Y=Y+1:GOSUB 440
360 NEXT I
370 A=FNR(8)+1:GOSUB 400:IF FNP(GX¥8-2,GY¥8)<>32 THEN A=
(A+3)MOD8+1:GOSUB 400
380 PUT SPRITE 4,(GX,GY),7,4:GOTO 300
390 *
400 GX=(((GX-16)+FNX(A)*8+200) MOD 200)+16
410 GY=(((GY-8)+FNY(A)*8+160) MOD 160)+8
420 RETURN
430 *
440 IF VPEEK(6146+X+32*Y)=32 THEN RETURN
450 PLAY "L6404GFEDC"
460 END
470 *
480 DATA 00,80,80,80,FF,FE,FC,00,C0,CF,CF,FF,FF,FF,FF,00
490 DATA 07,07,07,FF,FF,FF,7F,00,A5,7E,3C,FF,FF,3C,7E,A5


```


24

敵の攻撃をかわし、万里の長城を築け！

万里の長城

ところは中国西北部。北からの異民族の侵略は、ますます激しさを増していた。そこで時の朝廷は、異民族との境界線に防御壁を築くことになった。敵の攻撃は工事中も執ように繰り返される。さて、キミは万里の長城を築くことができるか？



PLAY




カーソルキーの上下左右キーを使って、画面上にあるブロックを画面中段までもって行って、“万里の長城”を完成させるゲームだ。3行の防御壁はすでにできており、つぎの4行目をブロックで並べると……バンザイ！ 1面クリアとなる。

ウム、簡単ではないか。最初のうちは敵の大砲攻撃も少ないし、3列完成されると、敵の攻撃をうけずにすむ柵ができるので安心……と思ったら話ががちがうじゃないか、敵の大砲によって縦に空白ができた防御壁から大砲の弾が入ってゲームオーバー！

PROGRAM

このゲームでは、万里の長城の1つの列のブロック数を配列の中に入れて、種々の処理をしている。そこで配列についての説明を少し。

たとえば、テストの点数の平均点を出す場合、10人の点数を P_1, P_2, \dots, P_{10} という変数にして、平均を $A = (P_1 + P_2 + \dots + P_{10}) / 10$ としてもいいけれど、この方法だと100人になったらたいへんだね。こんな時に配列を使うと便利だ。1番目の人の点数を $P(1)$ 、2番目を $P(2)$ ……という配列に振りわけ、平均点なら $S = 0$: FOR $I = 1$ TO 100 : $S = S + P(I)$: NEXT I $H = S / 100$ とすればずいぶん簡単になる。このようにカッコの中の添字によってそれぞれを区別できるのが配列だ。



実直に端から築いていくのがよいようだ

絶妙コントロールゲーム

```

100 '*** banri ***
110 '
120 COLOR 15,1:SCREEN 1,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DIM BL(24)
140 DEF FNR(A)=RND(1)*A:DEF FNB$(N)=STRING$(N,"□")
150 DEF FNP(X,Y)=VPEEK(6146+X+Y*32)
160 DEF FNX(S)=VAL(MID$("112221000",S+1,1))-1
170 DEF FNY(S)=VAL(MID$("100122210",S+1,1))-1
180 FOR I=1 TO 8
190   READ A$:S$=S$+CHR$(VAL("&H"+A$))
200 NEXT I:SPRITE$(1)=S$
210 '
220 CLS:FOR I=1 TO 24:BL(I)=3:NEXT I
230 LOCATE 1,6:PRINT FNB$(24)," ";FNB$(24)," ";FNB$(24):
PX=0:PY=22:LE=LE+1:LOCATE0,0:PRINT "PAGE ";LE:D$=STRING$(
24,"□")
240 '
250 LOCATE 1,21:PRINT D$:S=STICK(0):IF S=0 THEN 280 ELSE
LOCATE PX,PY:PRINT " "
260 X=PX:Y=PY:GOSUB 420:P=FNP(X,Y):IF P=32 THEN PX=X:PY=
Y ELSE IF S=1 AND P=219 THEN PX=X:PY=Y:GOSUB 380
270 PUT SPRITE 1,(PX*8+16,PY*8),9,1
280 IF DY=0 THEN IF FNR(80)/(LE*.2+1)<1 THEN DY=1:DX=FNR
(24)+1 ELSE 350
290 LOCATE DX,DY:PRINT " ":DY=DY+1
300 P=FNP(DX,DY):IF P=32 THEN 330
310 IF P<>219 THEN DY=0:GOTO 350
320 DY=0:LOCATE DX,5+BL(DX):PRINT " ":BL(DX)=BL(DX)-1:PL
AY "", "L6404BFC":GOTO 350
330 IF DY>12 THEN PLAY "L3203BAGFEDC":END
340 LOCATE DX,DY:PRINT "*" :PLAY "", "L6407C"
350 IF D$=SPACE$(24) THEN FOR I=0 TO 5:PLAY "L6404CDEFGA
B":NEXT:FOR W=0 TO 4000:NEXT:GOTO 220
360 GOTO 250
370 '
380 GOSUB 420:PLAY "L64N80":IF Y<>10 OR BL(X)>3 THEN LOC
ATE X,Y:PRINT "□":RETURN
390 PLAY "L804CL64FG":BL(X)=BL(X)+1:LOCATE X,5+BL(X):PRI
NT "□":A=((X-1)¥3)*3+1:IF BL(A)+BL(A+1)+BL(A+2)=12 THEN
LOCATE A,5:PRINT "㊦":MID$(D$,A,3)="   ":PLAY "L6405
CDEFGAB"
400 RETURN
410 '
420 A=FNX(S):B=FNY(S):X=X+A:Y=Y+B:IF ABS(X*2-25)>25 OR A
BS(Y-16)>6 THEN X=X-A:Y=Y-B
430 RETURN
440 '
450 DATA 5D,5D,49.3E,1C,1C,36,63

```


アリの軍団を越え、あそこまで行けるか？

25

横断大作戦


庭の片隅にアリの行列。そこへ1匹のハエが舞い落ちた。行列の向こうの犬のう〇こ(!)に降りるつもりが、あまりの空腹に力尽きたのだ。「なんとしてもたどりつかねば」が、ヘタするとアリどものエジキだ！ 嗚呼、ハエの運命や如何に！？



PLAY

蟻は大昔から行列ずき。今回も、なんと7列も作り、それぞれ交互に左右へ進んでいる。ドッドッド。



ハエは画面上方で歩きまわっている。アリの行列のすきまのところにきたら、急いで下むきのキーを押す！ やった。はじめの列はアリの密度が少ないので、難なくクリア。ドッドッド、下の方へ行くほど密度が濃くなってくるので、だんだん苦しくなり始める。でも大丈夫、1面突破。しかし2面は、さらにゴチャゴチャしてるじゃないか、オイ、渡れるかい？



PROGRAM

アリの行列を作成しているのは、220行からだ、まず220行で画面の幅(WIDTH)を設定。つぎにFOR~NEXT文で長さWDの文字列を作成し、その中でまたFOR~NEXT文でランダムな位置にアリを設定する。すきまがなくなるのを防ぐため、1つだけ行列の最後に空白を入れる。最下行の列だけは、特別にスキ間が1つしかできないようになっている。

さて、こうやってできた行列を、こんどは動かすわけだが、これは470行で行っている。210行1つで右に動かすのも左に動かすこともできる。それはP(0)に1が、P(1)に行列の長さ(WD)-1が入っているからだ。



いまだ！ このすき間に入りこめ！

絶妙コントロールゲーム

```

100 '*** oodan ***
110 '
120 CLEAR 1000:COLOR 15,4:SCREEN 1,0:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DEF FNR(X)=INT(RND(1)*X)
140 DEF FNP(X,Y)=VPEEK(6144+X+Y*32)
150 FOR I=1 TO 8
160   READ A$:S$=S$+CHR$(VAL("&H"+A$))
170 NEXT I:SPRITE$(0)=S$
180 FOR I=0 TO 4
190   READ P$(I)
200 NEXT I:PLAY P$(0):PL=3
210 SE=SE+1:WD=32-SE*2:WIDTH WD
220 FOR I=1 TO 7
230   A$(I)=SPACE$(WD-1)
240   FOR J=1 TO I*5
250     MID$(A$(I),FNR(WD-1)+1)="0"
260   NEXT J
270   A$(I)=A$(I)+" ":A$(7)=STRING$(WD,"0")+ " ":LOCATE 0
,I*3:PRINT A$(I);
280 NEXT I
290 LOCATE 0,0:PRINT "PAGE";SE;"LEFT";PL-1:E=1:TIME=0
300 P(0)=1:P(1)=WD-1:PX=5+SE:PY=1:ON INTERVAL=100/(SE*.5
+3) GOSUB 430:INTERVAL ON
310 '
320 PUT SPRITE 0,(PX*8,PY*8),15,0:C=FNP(PX,PY):IF C=219
THEN INTERVAL OFF:GOTO 400
330 S=STICK(0):IF S=5 THEN 370
340 DX=1-((PY¥3) MOD 2)*2:BEEP:PX=PX+DX
350 IF PX=32-SE OR PX=SE THEN PX=32-PX
360 GOTO 320
370 INTERVAL OFF:PY=PY+1:BEEP
380 C=FNP(PX,PY):IF PY=21 AND C=32 THEN PLAY P$(1):CLS:G
OTO 210
390 INTERVAL ON:GOTO 320
400 PLAY P$(2):SE=SE-1:PL=PL-1:IF PL<>0 THEN 210
410 END
420 '
430 PLAY P$(3)+MID$(P$(4),(E MOD 4)*4+1,4):GOSUB 470:LOC
ATE 0,E*3:PRINT A$(E):E=E+1-((E+1)¥8)*7
440 LOCATE 15,0:PRINT "TIME":INT(30-TIME/60):IF TIME>180
0 THEN TIME=0:GOTO 400
450 C=FNP(PX,PY):IF C=219 THEN 400 ELSE PLAY "L16":RETUR
N
460 '
470 A$(E)=RIGHT$(A$(E),P(E MOD 2))+LEFT$(A$(E),WD-P(E MO
D 2)):RETURN
480 '
490 DATA DB,E7,FF,E7,7E,3C,7E,A5
500 DATA L16,L6405CDEFGAB,L1603GFEDC
510 DATA 02L64,FR64ER64DR64CR64

```

26

悪党どもを撃ち殺せ!

人間ハンティング

開拓時代のアメリカ西部の町。荒くれどもが集まるバーでは、けんかは日常茶飯事だ。きょうもまた騒ぎが始まった。だがどうもいつもと勝手が違う。治まるどころかますます大混乱。ついに悪党どもが一般市民を人質に店にたてこもってしまった。



PLAY

6つの窓がある。そこに人物が現れるので、対応するキーを押せば、そいつは撃ち殺せる。だが思い出せ、人質がいたことを。一般市民を撃ってはいけないぞ。キミの標的は3人のガンマンだ。丸腰で手を上げている男と、娘(に見えるよな)が出たら、撃つ代わりに必ずスペースバーを押すこと。

正しくキーを押せば100点プラス。相手に先に撃たれたり、一般市民を撃ったりしたら人がひとり減ってしまう。また、一般市民が出たのにスペースバーを押さなかったら200点のマイナス。




何かに似てると思ったら、そうだ、モグラたたきだ! あれに複雑な要素が加わったわけだ。

PROGRAM

モグラたたき同様、それぞれの窓に対応するキーは変えることが可能(480行参照)。

また、このプログラムはメイン部分が非常に短いのが特徴だ。370行からあとはデータだし、120行から210行まではゲームを始めるにあたっての初期設定だし、220行から270行はゲームをはじめるために画面をつくる部分。つまりメインとなっている部分は、290~330行までのたった4行だ。

300行についていうと、変数Hはどの人物を出現させるか、Pはどの窓に出させるか、Cはどの色にするかを、それぞれ決定する。



窓に現れたのは……、敵だ! 撃て!!

反射神経養成ゲーム

```

100 '*** hunt-1 ***
110 '
120 SCREEN 1:WIDTH 29:KEY OFF:DEFINT A-Z
130 PRINT "LEVEL":PRINT:INPUT " 1 [ヤサイ] - 5 [ムスカシ]";L
E:IF LE<1 OR LE>5 THEN 120
140 COLOR 15,1:SCREEN 2,3:R=RND(-TIME):DEFFNR(A)=RND(A)*
A:OPEN "GRP:" AS #1
150 FOR I=0 TO 4:S$=""
160   FOR J=1 TO 32:READ A$:S$=S$+CHR$(VAL("&H"+A$)):NEX
T J
170   SPRITE$(1)=S$
180 NEXT I
190 FOR I=0 TO 5
200   READ I$(1)
210 NEXT
220 LF=3:LINE(0,8)-(255,8),11
230 FOR I=0 TO 1
240   FOR J=0 TO 2
250     FOR K=0 TO 3:LINE(28+72*J+K,20+80*I+K)-(67+72*J-
K,83+80*I-K),10,B:NEXT K
260   NEXT J
270 NEXT I
280 '
290 LINE(8,0)-(200,7),1,BF:PRESET(8,0):PRINT #1,"SCORE "
;SC;" LEFT";LF
300 H=FNR(5):P=FNR(6):C=FNR(13)+2:PUT SPRITE 1,(32+72*(P
MOD3),80*(P/3)+48),C,H
310 A$=INKEY$:WA=WA+1:IF WA<(6-LE)*50 AND A$="" THEN 310
ELSE WA=0
320 IF (A$=I$(P) AND H<3) OR (A$="" AND H>2) THEN PLAY
"L6405C":SC=SC+100:GOTO 290
330 IF NOT(A$=I$(P) AND H>2) AND H>=3 THEN SC=SC-200:GOT
O 290
340 LF=LF-1:IF LF>=1 THEN PLAY "L6403BAGFEDC":FOR I=0 TO
1000:NEXT:GOTO 290
350 PLAY "L102D":SCREEN 1:PRINT "Score ";SC:END
360 '
370 DATA 01.01.01.01.01.03.07.07.01.00.00.00.00.00.00.00
380 DATA 07.07.3F.07.0F.07.03.0F.FF.FF.0F.0F.0F.05.0F.0F
390 DATA F0.FC.F0.F0.E0.C0.F8.FC.FE.F7.F3.F0.50.F0.F0.F0
400 DATA 00.00.00.00.00.00.00.00.40.E0.F0.98.04.00.00.00
410 DATA 03.03.0F.03.03.03.01.0F.3F.63.63.33.2B.13.07.03
420 DATA C0.C0.F0.C0.C0.C0.80.F0.F8.FC.EC.EC.EC.EC.FC.F0
430 DATA 03.03.03.07.03.01.01.03.07.1F.3B.03.03.07.0F.0F
440 DATA E0.F0.F0.F0.D0.90.90.D8.CC.C4.C0.C0.E0.F0.F0.F0
450 DATA 03.03.0F.63.63.61.63.7F.3F.07.07.07.07.0A.0F.0E
460 DATA E0.E0.F8.E3.E3.C3.E3.FF.FE.F0.F0.F0.F0.A8.F8.78
470 '
480 DATA A.S.D.Z.X.C

```

ニブイヤつはモグラにまでナメられるぞ!

27

モグラくんの1日

懐かしのオールドゲームシリーズ。このゲームはかつてはウサ晴らしの代名詞であり、ゲームセンターで目を血走らせて熱中するサラリーマンの姿は1つの風俗だった。時は移り MSX でモグラたたき。ウサ晴らしというより、反射神経養成用?



PLAY




ルールはいまさら言うまでもなからうが、一応いっておくと、穴からモグラが顔を出したら、その穴に対応するキーを押せばよいのである。穴は9か所。キーはT・Y・U・H・J・B・N・M。注意してほしいことは **CAPS LOCK** キーのロックをかけて、大文字モードになっていることを確認すること。さもないといくら押してもミスと見なされる。

正しいキーを押すと+100点、まちがえると-100点。はじめのうちはモグラは長い間顔を出しているの、なあんだとタカをくくるかもしれないが、そのうちすぐに引っこんでしまうようになる。モグラの動きについていけなくなったら、まちがえても何もなくても同じなので、どのようにするかはキミの自由だよ。

PROGRAM

穴に対応するキーは簡単に変えられる。450行にデータとして与えてあるのでここを変えればよい。たとえばむずかしくするのならキーの配置をまったく逆にすればよいし、やさしくするのなら、極端な話、どの穴もすべて1つのキーに対応させればよい。

モグラが顔を出している時間がしだいに短くなるのは、キー入力待ちを **FOR~NEXT** ループの中に入れ、ループを回る回数を時間の関数で決めているからである(310行)。



モグラの表情見てよ、完全にバカにしてるよな

反射神経養成ゲーム

```

100 '*** mogura ***
110 '
120 COLOR 1,2:SCREEN 0:WIDTH 39:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DEF FNR(A)=RND(1)*A
140 INPUT "LEVEL [ 1(EASY) - 5 ]";LE:IF ABS(LE-3)>2 THEN
140 ELSE SCREEN 2,3
150 FOR I=1 TO 32
160   READ A$:S$=S$+CHR$(VAL("&H"+A$))
170 NEXT I:SPRITE$(1)=S$
180 FOR I=0 TO 8
190   READ A$(I)
200 NEXT I
210 OPEN "GRP:" AS #1
220 ON INTERVAL=3600 GOSUB 370:TIME=0:INTERVAL ON
230 FOR J=0 TO 2
240   FOR I=0 TO 2
250     A=I+J*3:X(A)=I*56+40:Y(A)=J*48+24:CIRCLE (X(A)+1
250     5,Y(A)+22),20,1,...5
260     PAINT (X(A)+15,Y(A)+22),1
270   NEXT I
280 NEXT J
290 '
300 N=FNR(9):PUT SPRITE 1,(X(N),Y(N)),FNR(13)+3,1
310 FOR TT=0 TO (3600-TIME)¥(LE+5):A$=INKEY$:IF A$="" TH
310 EN NEXT TT:PO=PO-50:PLAY "L6403BCAC":GOTO 330
320 IF A$=A$(N) THEN PO=PO+100:FOR I=0 TO 6:BEEP:NEXT I
320 ELSE PO=PO-50:PLAY "L6403GFEDC":FOR I=0 TO 280:NEXT I
330 PUT SPRITE 1,(0,209),1,1
340 LINE(0,0)-(255,7),2,BF:PRESET(0,0):PRINT #1,USING "S
340 core #####";PO:PRESET(150,0):PRINT #1,USING"TIME  #####
340 #":360-TIME¥10
350 GOTO 300
360 '
370 INTERVAL OFF
380 PRESET (80,70):PRINT #1,"GAME OVER":PLAY "L3204GRL32
380 DR64DR8D#RDR2F#RG":FOR I=0 TO 2500:NEXT I
390 SCREEN 1:LOCATE8,9:PRINT"GAME OVER":LOCATE5,14:PRINT
390 "YOUR SCORE ";PO
400 END
410 '
420 DATA 3F,7F,E3,EF,DE,FE,FF,FB,FC,FF,3F,87,CF,7F,1F,1F
430 DATA FC,FE,C7,F7,7B,7F,FF,DF,3F,FF,FC,E1,F3,FE,F8,F8
440 '
450 DATA T,Y,U,G,H,J,B,N,M

```

28

大蛇よ、パワーマークをねらえ!

ヘビ使い




アクションゲームは、ゲーム内容の単純さに比例して興奮度も増す。これはゲーム哲学の公理であ〜る。そこで登場するのがこのヘビ使いだ。画面の中をクネクネと伸びるヘビが、パワーマークを食べる。単純なストーリーだから味わえる興奮!

PLAY

まず、ヘビはカーソルキーの上下左右で操作する。画面の4辺にヘビがぶつくと、それだけでゲームオーバーなので注意。

画面にはボツンと星のようなマークがでるが、何を隠そうこれがパワーマークなのだ。このパワーマークをヘビの頭で取ると500ポイントアップ。さあ、PLAYだ。おっと、画面右上にパワーマークが出たゾ。伸びるヘビ、ねらうヘビというわけでバクリ。つぎの左下もバクリ。と、だんだんヘビの尻尾が長くなってきたぞ! ウーム、無念、結局自分の尻尾をヘビが食べてしまった。ヘビってバカね……。




再プレイ…… [SPACE BAR]
 止め方…… [CTRL] + [STOP]

PROGRAM

このプログラムでは2か所、VPEEK文を使っている。1つは、320~330行のヘビの頭の位置のデータを取りこむもので、もう1つは、410行目にある。この410行目のVPEEK文は、パワーマークが乱数を用いて置かれているため、ヘビの上やカベに乗ることを防ぐためのものだ。

ヘビの尻尾の処理は、TX(1000), TY(1000)という大きな配列のうち、ヘビの長さの値で行っている。たとえば、ヘビの長さが10とすると、TX(TL), TY(TL)が尻尾の位置を表すから、毎回そこを消せばよいことになる。



パワーマークが出た、急いで食うのだ!

反射神経養成ゲーム

```

100 '*** snake ***
110 '
120 COLOR 15,4:SCREEN 0:WIDTH 39:KEY OFF
130 DEFINT A-Z:R=RND(-TIME):DIM TX(999),TY(999)
140 DEF FNR(X)=RND(1)*X:DEF FNP(X,Y)=VPEEK(X+40*Y+1)
150 DEF FNX(S)=VAL(MID$("11211101",S,1))-1
160 DEF FNY(S)=VAL(MID$("01112111",S,1))-1
170 '
180 XX=5:YY=5:OX=5:OY=5:DX=1:DY=0:TL=0:TS=0:L=5:PO=0:TIME
E=0
190 FOR I=0 TO 5
200   TX(I)=OX:TY(I)=OY
210 NEXT I
220 CLS
230 FOR I=0 TO 22
240   LOCATE 0,I:PRINT "X";
250   LOCATE 38,I:PRINT "X";
260 NEXT I
270 LOCATE 0,0:PRINT STRING$(38,"X");:LOCATE 0,22:PRINT
STRING$(38,"X");
280 GOSUB 410
290 '
300 S=STICK(0):IF S MOD 2<>0 THEN DX=FNX(S):DY=FNY(S)
310 XX=XX+DX:YY=YY+DY
320 IF FNP(XX,YY)=42 THEN PO=PO+500:GOSUB 410:GOTO 340
330 IF FNP(XX,YY)<>32 THEN 480
340 GOSUB 470:LOCATE OX,OY:PRINT "□";
350 LOCATE TX(TL),TY(TL):PRINT " ";
360 LOCATE XX,YY:PRINT "◆";
370 OX=XX:OY=YY:TL=(TL+1) MOD 1000
380 LOCATE 25,23:PRINT USING "#####":PO+(TIME¥60)*10;
390 GOTO 300
400 '
410 XA=FNR(39):YA=FNR(23):IF FNP(XA,YA)<>32 THEN 410
420 LOCATE XA,YA:PRINT "*":BEEP
430 FOR I=1 TO 5
440   GOSUB 470:L=L+1
450 NEXT I:RETURN
460 '
470 TX((TL+L) MOD 1000)=OX:TY((TL+L) MOD 1000)=OY:RETURN
480 PO=PO+(TIME¥60)*10:IF HI<PO THEN HI=PO
490 PLAY "O4L64BAGFEDC":LOCATE 0,23:PRINT USING "HighSco
re #####":HI;
500 IF INKEY$<>" " THEN 500 ELSE 180

```

29

アクション陣取り、かけひきの妙

ベクトルゲーム

日ごろから鈍いやつといわれつけているキミは、このふたり用ゲームで名誉ばん回を図れ。反射神経とかけひきの能力の差が歴然とするこのゲームは、どっちが鈍いかに決着をつけてくれる。こうしてキミは自他ともに鈍いことを認められるのだ。



PLAY


画面の左上にプレイヤー1が、右下にプレイヤー2が陣取る。1は本体のカーソルキーで、2はジョイスティックでそれぞれのベクトルを操作する。ベクトルは画面を上下左右に進みながら軌跡を残し、へびのように細長く自分のからだを伸ばしていく。2人のうち先に壁や、自分か相手のベクトルにぶつかったほうが負けてある。

したがって勝負は、いかに相手を狭い領域へ追い込むか、というかけひきにかかっている。もちろん自分のベクトルを思い通りに操作する技術を身につけねばならない、なんてそれ以前の問題だぞ。



PROGRAM

基本的なプログラムはへび使いと変わらないが、2つのベクトルを動かすために、280~300行のFOR~NEXTループで2回まわっている。ふたりのプレイヤーは変数Pで表される。Pが0のときはプレイヤー1、Pが1のときはプレイヤー2だ。

各ベクトルの頭の位置はXX(P), YY(P)で、方向はCH(P)で表される。CH(P)が0, 1, 2, 3のときそれぞれ上, 左, 下, 右を向いている。CH\$(CH(P))という文字が方向を示している。430行をいじればスタート時のベクトルの位置と方向を変えられる。スタート直後にぶつかるようにセットする、なんてのもスリリングだ。



プレイヤー2……ジョイスティックを使用
再プレイ……RUN (F.5)



しまった! 包囲された!

反射神経養成ゲーム

```

100 '*** vector ***
110 '
120 COLOR 15,4:SCREEN 0:WIDTH 40:KEY OFF
130 DEFINT A-Z:DEF FNP(X,Y)=VPEEK(X+Y*40)
140 DEF FNP(X,Y)=VPEEK(X+Y*40)
150 DEF FNX(S)=VAL(MID$("11211101",S,1))-1
160 DEF FNY(S)=VAL(MID$("01112111",S,1))-1
170 FOR I=0 TO 5
180   READ CH$(I)
190 NEXT I
200 FOR I=0 TO 1
210   READ XX(I),YY(I),OX(I),OY(I),DX(I),DY(I),CH(I)
220 NEXT I
230 FOR I=0 TO 23
240   LOCATE 0,1:PRINT "□":LOCATE 39,1:PRINT "□":
250 NEXT I
260 LOCATE 0,0:PRINT STRING$(39,"□");:LOCATE 0,22:PRINT
STRING$(39,"□");
270 '
280 FOR P=0 TO 1
290   GOSUB 330
300 NEXT
310 GOTO 280
320 '
330 S=STICK(P):IF S MOD 2<>0 THEN DX(P)=FNX(S):DY(P)=FNY
(S)
340 XX(P)=XX(P)+DX(P):YY(P)=YY(P)+DY(P)
350 CH(P)=DX(P)+DY(P)+ABS(DX(P))*2+ABS(DY(P))
360 IF FNP(XX(P),YY(P))<>32 THEN LOCATE 5,24:PRINT "Play
er ";2-P;"ノカチテス!!":PLAY "05L64CDEFGAB":END
370 LOCATE OX(P),OY(P):PRINT CH$(P+4);
380 LOCATE XX(P),YY(P):PRINT CH$(CH(P));
390 OX(P)=XX(P):OY(P)=YY(P)
400 BEEP:RETURN
410 '
420 DATA ⊕,⊕,⊕,⊕,*,#
430 DATA 5,5,5,5,0,1,0,34,18,34,18,0,-1,3

```


30

赤あげて、白あげないで赤さげない 旗上げゲーム

諸君は、あのなつかしき「旗上げゲーム」を憶えてるかな？
そう、昔テレビでよくやってた「赤あげて、白さげて」というア
シだ。やる者、見る者ともに興奮のルツボと化させた(!)この
傑作ゲームを、いまここに謹んで贈ろう。



PLAY

やり方は単純明快。「アカアゲテ」と言われたら赤
旗を上げ、「シロアゲテ」と言われたら白旗を上げれ
ばいいのだ。旗の上げ下げはファンクションキーの
1~4でやる。各キーが何色の上下に対応するかは
画面上に表示される。

最初に1~5のレベルを選ぶ。1回のゲームで20
回の「アカアゲテ」他の指示があるが、「20レベル
数」の回数を成功したら、1ランク上のレベルに移る
ことができるようになっている。

PROGRAM

このゲームは、この本のプログラム中、唯一、フ
ァンクションキー割り込みの ON FUNCTION という文を使っている。が、これも、いままでの割り込
みと同じに考えてよい。違うのは、ファンクション
キーは多数個あるので、その番号に対応して飛び先
を変えられること。つまり、150行で ON KEY
GOSUB -, -, -, -, となっている“-”の部分に各
処理の頭の番号で指定するというわけだ。これは、
一般文中の ON “変数” GOSUB -, -, ……の“変数”
の部分がファンクションキーの番号に対応すると思
えばよい。

あとは、各キーの割り込みを許可する命令の210
行 KEY(n) ON を実行すれば、押されたファンク
ションキーの番号にそった処理へ分岐する。

Score = 30

カイスウ = 20

アカアゲテ
赤い旗
白い旗
コンピュータの答え
アホ!!
ファンクションキーに対応する命令
アカアゲル アカサゲル シロアゲル シロサゲル

F1 F2 F3 F4

再プレイ……

止め方…… CTRL + STOP

Score = 20 カイスウ = ?

シロアゲテ

アホ!!

アカアゲル アカサゲル シロアゲル シロサゲル

あつ、まちがえた

反射神経養成ゲーム

```

100 '*** hataage ***
110 '
120 COLOR 15,1,4:SCREEN 1:WIDTH 29:R=RND(-TIME)
130 DEFINT A-Z:INPUT "LEVEL":LV:SCREEN 2,1:OPEN "GRP:" A
S #1:GOSUB 410
140 FOR I=1 TO 4:READ W$(I),P$(I):PRESET (-50+I*60,180):
PRINT #1,W$(I)+"ル":NEXT
150 ON KEY GOSUB 290,300,320,330
160 '
170 SC=0:R=0:W=0:GOSUB 350:PLAY "L64"
180 HT=0
190 FOR K=1 TO 20
200   GOSUB 370
210   FOR I=1 TO 4:KEY(I) ON:NEXT I
220   FOR I=0 TO 1800-30*.5*K*LV:NEXT I
230   FOR I=1 TO 4:KEY(I) STOP:NEXT I
240   GOSUB 380:GOSUB 400
250 NEXT K
260 IF HT>20-LV THEN FOR W=0 TO 4000:NEXT:LV=LV+1:GOTO 1
80
270 LINE (100,130)-(148,140),1,BF:IF INKEY$((">CHR$(13) TH
EN BEEP:GOTO 270 ELSE 170
280 '
290 RF=1:GOTO 310
300 RF=0:GOTO 310
310 LINE (X-8,Y-10)-(X-16,Y-18+16*RF),1:PUT SPRITE 1,(X-
32,Y-10-RF*16),9,1:LINE (X-8,Y-10)-(X-16,Y-2-16*RF):RETU
RN
320 WF=1:GOTO 340
330 WF=0:GOTO 340
340 LINE (X+8,Y-10)-(X+16,Y-18+16*WF),1:PUT SPRITE 2,(X+
18,Y-10-16*WF),15,1:LINE (X+8,Y-10)-(X+16,Y-2-16*WF):RET
URN
350 X=130:Y=100:CIRCLE (X,Y-20),5,11:LINE (X-8,Y-12)-(X+
8,Y-4),7,BF:LINE (X-6,Y-4)-(X-3,Y+8),.BF:LINE (X+3,Y-4)-
(X+6,Y+8),.BF:GOSUB 300:GOSUB 330
360 LINE (10,1)-(160,9),1,BF:PRESET (10,1):PRINT #1,"Sco
re=":SC:"カイスウ=":K:RETURN
370 A=INT(4*RND(1))+1:PLAY P$(A):PRESET (100,20):PRINT #
1,W$(A)+"テ":IF A<=2 THEN R=1-INT(A/2):RETURN ELSE W=1-IN
T(A/4):RETURN
380 LINE (100,130)-(148,140),1,BF:PRESET (100,130):IF (R
F=R) AND (WF=W) THEN PRINT #1,"GOOD !":PLAY "O6AAA":HT=H
T+1:SC=SC+10 ELSE PRINT #1,"アホ !!":PLAY "O3AAA"
390 GOSUB 360:RETURN
400 LINE (100,20)-(200,50),1,BF:RETURN
410 FOR I=1 TO 8:READ A$:S$=S$+CHR$(VAL("&H"+A$)):NEXT:S
PRITE$(I)=S$:RETURN
420 DATA 18,3C,7E,FF,FF,7E,3C,18
430 DATA アカアケ~,04GGB,アカサケ~,04GGD,シロアケ~,04EEB,シロサケ~,04EED

```


31

カール・ルイスに挑め

100m走



「日本のみなさん今晚は、こちらロサンゼルス市のオリンピック陸上競技会場です。いま、こちらでは、世界一速い男を決定する、短距離の花形、100m走の決勝が行われようとしております!!」な～んてアナウンスはないけど気分は世界記録!

PLAY




ゲームセンターをスポーツセンターにした、あのハイパーオリンピックの基本競技100m走を、MSXで再現。画面に「ヨーイ」と表示されたら、カーソル左右移動キーに指をのせる。「ドン」の字とともに響くピストル音。あとはもうゴールに着くまで必死にキーをたたきまくるのみ、というソーカイさ。

ただし、このゲームはただ単に速くキーをたたきただけではダメで、左右のキーを必ず交互にたたかねばならないというのがむずかしい点だ。さあ、世界記録(作者の記録)9.67秒をめざして走れ走れ!!

PROGRAM

交互にキーを押さなければ、ランナーが進まないようにしているのが、200行の「ON F GOTO」と、飛び先の210, 220行。150行ではじめにF=1が設定されているから「ON F GOTO」で210行に行く。そこで右向きキーが押されていればFを2にして1m走る。今度は「ON F GOTO」でFが2だから220行へ。そこで左向きキーが押されていればFを1にして1m走る。そしてまた「ON F GOTO」へ…。つまりFという変数が、前回どちらのキーを押したかを覚えているワケ。

ゴールの判定は、Lが110(所要時間が10秒くらいになるよう、この値とした)になったかで判定し、タイム値 TIME から時間を算出、表示する。



この軽快な走りっぷり。また世界新記録更新かな

単純興奮アクションゲーム

```

100 ' *** r100m ***
110 '
120 COLOR 15,4,7:SCREEN 2,3,0:KEY OFF
130 DEFINT A-Z:OPEN "GRP:" AS #1:GOSUB 280:HI!=11!
140 CLS:LINE (0,170)-(250,170),2:LINE (233,150)-STEP (7,
5),15,BF
150 TIME=0:L=0:F=1
160 PRESET (20,20):PRINT #1,"ヨ-イ ":PUT SPRITE 1,(0,136),
11,F:PLAY "O5L16EFGABBBB"
170 IF TIME<120 THEN 170 ELSE PRESET (20,30):PRINT #1,"ト
ン ":RESTORE 400:FOR I=0 TO 13:READ D:SOUND I,D:NEXT:TIM
E=0
180 '
190 X=2*L:PUT SPRITE 1,(X,136),11,F
200 IF L=110 THEN 230 ELSE ON F GOTO 210,220
210 IF STICK(0)=3 THEN F=2:L=L+1:GOTO 190 ELSE 210
220 IF STICK(0)=7 THEN F=1:L=L+1:GOTO 190 ELSE 220
230 TM!=TIME/60:PRESET (20,40):PRINT #1,USING "###.##秒"
:TM!:SOUND 7,&HF8:PLAY "L16GGR16GR64GR64GGGG"
240 IF HI!>TM! THEN HI!=TM!
250 PRESET (20,60):PRINT #1,USING "High Score = ###.##秒
":HI!
260 IF INPUT$(1)<>" " THEN PUT SPRITE 1,(0,209):GOTO 260
ELSE GOTO 140
270 '
280 FOR I=1 TO 2
290   S$=""
300   FOR J=0 TO 31
310     READ A$:S$=S$+CHR$(VAL("&H"+A$))
320     NEXT J:SPRITE$(I)=S$
330 NEXT I:RETURN
340 '
350 DATA 00,01,01,00,00,0F,1B,1B,1B,03,13,3E,6C,00,00,00
360 DATA E0,F8,E0,E0,C0,E0,EC,B8,B0,80,C0,60,30,30,30,18
370 DATA 00,01,01,01,00,01,03,03,03,03,01,03,06,0C,0C,04
380 DATA E0,F8,E0,E0,C0,C0,C0,C0,F0,80,C0,60,30,60,C0,80
390 '
400 DATA 0,0,0,0,0,0,21,247,16,0,0,100,60,0

```

ゴツツアandes

32

押しすもう



「ひがああ～しいい～，わかあしまあずう～，にい～しいい～，ちよあの～ふうじい～」いけ，それ押しせ，トンコトコトン，と机をたたいてやる紙すもうの興奮。キミは知っているか？ それをコンピュータでやるとこうなるのであります。




PLAY

技はとにかく「押し」しかないのだ。そのためにひたすらスペースバーをたたきまくる体力＝勝利というゲームなのだ。

相手はコンピュータだ。人間が負けてなるものか、トントンドンドン押しせ、押しせ、押しすんだ。いけ、トトンドド、おっと相手の足が土俵にかかる。やった！ 押し出した(これしかないんだから)。時間は？ 12.75秒、まずまずですな。ゴツツアんです。それにしても…スペースバーが心配だナァ。

PROGRAM

プログラムを走らせると力士どうしがスーッと近づいて、その後で押し合いが始まるが、近づくまではスペースバーを叩いても反応しない。これは、トリガーの割り込み許可をスプライト割り込みの中で行っているからだ(440行)。つまり、初め、ふたりの力士は離れているから240行で決められた DX, XI に従って近づいてくる。そしてふたりがぶつかると、スプライトの割り込みがかかり、440行で DX と XI を同じ値-2にする。これでふたりは同方向の左へ動き始めるのだ。それと同時にトリガー割り込みを許可してプレイヤーから入力を受け付ける。あとは、トリガー割り込みがかかるたびに400行によって右へ動く。これで、割り込み回数が多いほど、プレイヤーの押しが強くなることになるわけ。



土俵ぎわまで追いつめたが、ゼイゼイ、疲れた

単純興奮アクションゲーム

```

100 ' *** sumo ***
110 '
120 COLOR 15,10,10:SCREEN 2,3,0:OPEN "GRP:" AS #1
130 DEFINT A-Z:HI!=20
140 FOR J=0 TO 3
150   S$=""
160   FOR I=1 TO 32
170     READ A$:S$=S$+CHR$(VAL("&H"+A$))
180   NEXT I
190   SPRITE$(J)=S$
200 NEXT J
210 CLS:CIRCLE (128,130),80,4,...5
220 LINE (100,120)-(100,140):LINE (156,120)-(156,140)
230 ON STRIG GOSUB 400:ON SPRITE GOSUB 440:SPRITE ON
240 X(0)=80:Y=85:X(1)=130:DX=2:X1=-2:TIME=0
250 FOR I=0 TO 1
260   PUT SPRITE I*2,(X(1),Y),6+I*3,I*2
270   PUT SPRITE 1*2+1,STEP(0.32),6+I*3,I*2+1
280 NEXT I
290 X(0)=X(0)+DX:X(1)=X(1)+X1
300 IF X(0)>=38 AND X(1)<=175 THEN 250
310 '
320 STRIG(0) OFF:TM!=TIME/60
330 PRESET (50,10):PRINT #1,USING "TIME ####.## 秒":TM!
340 PRESET (50,29):IF X(0)>48 THEN PRINT #1,"オエノ カチ!!":
IF HI!>TM! THEN HI!=TM! ELSE PRINT #1,"ワシノ カチ!!"
350 PRESET (50,50):PRINT #1,USING "High Score=####.## 秒
";HI!
360 IF INKEY$(<>CHR$(13)) THEN 360
370 FOR I=0 TO 3:PUT SPRITE I,(0,209):NEXT I
380 GOTO 210
390 '
400 RESTORE 550:FOR S=0 TO 13:READ D:SOUND S,D:NEXT S
410 X(0)=X(0)+3:X(1)=X(1)+3
420 RETURN
430 '
440 SPRITE OFF:DX=X1:STRIG(0) ON:RETURN
450 '
460 DATA 3F,0F,0F,1F,1F,3F,37,07,07,0F,3F,7F,FF,FF,FE,FF
470 DATA 80,C0,C0,E0,E0,C0,C0,80,00,C0,E0,F1,D9,E3,FF,7F
480 DATA FF,AA,FF,55,FF,FF,7F,3F,0F,01,03,07,0F,1F,1F,1F
490 DATA 7F,81,FC,54,F0,F0,C0,F8,FC,FC,FC,F8,F0,C0,E0,F0
500 DATA 01,03,03,07,07,03,03,01,01,03,07,8F,9B,C7,FF,FE
510 DATA FC,F0,F0,38,B8,BC,6C,E0,F0,F8,FC,FE,FF,BF,7F,FF
520 DATA FE,81,3F,2A,1F,0F,0F,1F,3F,3F,3F,1F,0F,03,07,0F
530 DATA FF,55,FF,AA,FF,FF,FE,FC,F0,C0,C0,E0,E0,F0,F8,F8
540 '
550 DATA 30,1,30,1,30,1,0,24,16,16,16,0,15,0

```


スペースバーをたたきまくる腕力ゲーム？

33

自動車レース

毎度おなじみ自動車レース。パソコンが操るマシンとキミのマシンの一騎打ちだ。といってもプレイヤーは、カタカタとスペースバーを押すだけ。いたって単純だけど、これが思わず熱くなったりして……。こんな短いプログラムでも十分遊べるゾ。



PLAY




プログラムを RUN させてみる。第一印象は「おっ、なかなかウツクシイな〜」。グレーの道路に、左側のオレンジの車が自分、右のグリーンのが相手車だ。と、見とれる間もなくスタートの合図。カタカタとスペースバーを一心にたたき。ふふふ、なんだい楽勝だぜい。圧倒的な差でゴールに突入、第2レースへと進む。

しかし、敵の車も面が進むごとにスピードが速くなってきている。ウヌー、カタカタカタ……(スペースバーを押す音)。しかしこんなにぶったたいて、スペースバーがもつかしらん？

PROGRAM

スペースバーを押したときにだけ特別な処理を行うのに便利なのが、ON STRIG GOSUB だ。こちらの車はスペースバーを押しただけ進むし、相手の車は INTERVAL を使って、一定時間に進む量を決めてある。この両者によってそれぞれの車の動きをコントロールしているわけだ。

<改良法> コンピュータの車が1度に動く量を変えたいとき、420行の $CY=CY-8$ を変えればよいし、プレイヤー側を変えたいときは480行の $PY=PY-8$ を変えればよい。車の色を変える場合は、コンピュータの車なら430行の7の数字を、自分の車なら490行の9の数字を変えてみよう。



圧倒的な差で第1レースは勝利

単純興奮アクションゲーム

```

100 ' *** car-r ***
110 '
120 COLOR 3,1:SCREEN 2,2,0:KEY OFF
130 DEFINT A-Z:OPEN "GRP:" AS #1
140 ON STRIG GOSUB 480:PLAY "S14M20"
150 S$=""
160 FOR I=1 TO 32
170   READ A$:S$=S$+CHR$(VAL("&H"+A$))
180 NEXT
190 SPRITE$(1)=S$
200 '
210 FOR I=0 TO 63 STEP 63
220   LINE(99+I,0)-(102+I,191),10,BF
230 NEXT I
240 LINE(103,0)-(161,191),14,BF
250 FOR I=0 TO 160 STEP 32
260   LINE(130.4+I)-(133.27+I),15,BF
270 NEXT I
280 LINE(99,176)-(165,176),3
290 PRESET(170,3):PRINT #1,"コ`ル"
300 PRESET(170,173):PRINT #1,"スタート"
310 SE=SE+1:PY=176:CY=176:TIME=0
320 ON INTERVAL=30/(SE*.3+2) GOSUB 420
330 LINE(8,3)-(64,11),1,BF
340 PRESET(8,3):PRINT #1,USING "PAGE ##":SE
350 PUT SPRITE 1,(107,CY),9,1
360 PUT SPRITE 2,(141,CY),7,1
370 PLAY "L404AR4AR4AR405A."
380 FOR I=0 TO 4300:NEXT I:INTERVAL ON:STRIG(0) ON
390 '
400 GOTO 400
410 '
420 CY=CY-8
430 IF CY>=0 THEN PUT SPRITE 2,(141,CY),7,1:RETURN
440 STRIG(0) OFF
450 PLAY "L1D":SCREEN 0:PRINT "GAME OVER"
460 END
470 '
480 PY=PY-8
490 IF PY>=0 THEN PUT SPRITE 1,(107,PY),9,1:RETURN
500 INTERVAL OFF:STRIG(0) OFF
510 PLAY "07L32BCBCBCBCBC":FOR I=0 TO 4000:NEXT
520 RETURN 310
530 '
540 DATA 03,07,67,7F,7F,67,07,07,06,E6,E7,FF,FF,E7,E5,05
550 DATA C0,E0,E6,FE,FE,E6,E0,E0,60,67,E7,FF,FF,E7,A7,A0

```


34

大地に眠るダイヤを掘りおこせ!

ダイヤモンド探し



永遠の輝きを放つダイヤモンド。その神秘的な輝きを求めて、キミはダイヤ探しへと旅立った。地底深くにダイヤが眠っているという情報を得たからだ。見事ダイヤを発見すれば、キミは莫大な富を手にするが、一步まちがえば……死が待っている。



PLAY

このゲームでは、なによりもまず掘って掘って掘りまくることだ。RUNしたら、ココ掘れワンワン、というわけでスペースバーを押して掘りまくる。

さ〜て、掘った後は土砂崩れと不発弾との闘いだ。この土地は極めて地盤が弱い。掘っている途中で手を休めると、すぐに土砂崩れで穴が埋まって上の方へ押しもどされてしまうのだ。し、しかも、ダイヤが埋まっていない場所には不発弾が埋まっている。それに触れると……ゲームオーバー! クソ、ダイヤはどこだ! エッサ、ホイサと掘り進むと、アッタゾ、ダイヤ発見! 流れる音楽が祝福を……。




PROGRAM

このゲームでも、穴を掘るために例のトリガー割り込みを使っている。(割り込みルーチンである) 350行以下でプレイヤーを下げたり、プレイヤーの穴の中での深さに応じて不発弾を発生させ、触れたときには爆発させゲームオーバーにしたり、などの処理を行っている。

プレイヤーの位置は、PX、PYに入っている。ダイヤのX座標はGXに入っているから、これら2つのかね合いで、この割り込みルーチン内の場合分けを行っているのだ。

また、このゲームでは290~310行で土砂崩れの処理のためにインターバルの割り込みも行っている。



ギョッ! 不発弾を掘りあてちゃった

単純興奮アクションゲーム

```

100 '*** diamond ***
110 '
120 COLOR 15,7:SCREEN 2,2:KEY OFF:DEFINT A-U:R=RND(-TIME
)
130 DEFFNB(L,C,Y)=SGN(FIX((L+1)/(ABS(C-Y)+1)))*(Y-C)+((Y
-C)¥(L+1))*L+C
140 FOR J=0 TO 2
150   S$=""
160   FOR I=1 TO 32
170     READ A$:S$=S$+CHR$(VAL("&H"+A$))
180   NEXT
190   SPRITE$(J)=S$
200 NEXT
210 OPEN "GRP:" AS #1
220 SE=1:PO=0:ON STRIG GOSUB 350:PLAY "L6402"
230 PX=0:PY=16:ON INTERVAL=200/(SE*.5+2) GOSUB 290
240 LINE (0,32)-(240,176),10,BF:GX=INT(RND(1)*15)*16:GY=
176:TIME=0:PUT SPRITE 1,(0,209):INTERVAL ON
250 STRIG(0) ON:PUT SPRITE 0,(PX,PY),15,0
260 IF PY=16 THEN S=STICK(0):IF S=3 OR S=7 THEN PX=FNB(1
12,112,PX-(S-5)*8)
270 IF PY<80 THEN PUT SPRITE 2,(0,209)
280 FOR W=0 TO 50:NEXT:GOTO 250
290 GOSUB 320:PRINT #1,USING "TIME ### SCORE #####":IN
T(100-TIME/60):PO
300 IF TIME>=6000 THEN 330
310 IF PY=16 THEN RETURN ELSE PLAY "A":PY=PY-16:PUT SPRI
TE 0,(PX,PY),15,0:LINE (PX,PY+16)-STEP(16,16),10,BF:STRI
G(0) ON:RETURN
320 STRIG(0) OFF:LINE (40,0)-(240,8),7,BF:PRESET (40,0):
RETURN
330 PLAY "05GFEDC":GOSUB 320:PRINT #1,"GAME OVER. Score
---":PO
340 IF INKEY$<>CHR$(13) THEN 340 ELSE 220
350 INTERVAL OFF:PLAY "C":PY=FNB(64,96,PY+16)
360 IF PY=112 AND PX<>GX THEN PUT SPRITE 2,(PX,128),9,2
370 IF PY>=128 AND PX<>GX THEN 330
380 IF PY=160 AND PX=GX THEN PLAY "CDEFGAB":PUT SPRITE 1
,(GX,GY),15,1:PO=PO+(6000-TIME)/2:FOR W=0 TO 2000:NEXT:S
E=SE+1:GOTO 230
390 PUT SPRITE 0,(PX,PY+4),14,0:LINE (PX,PY)-STEP(16,16)
,7,BF
400 INTERVAL ON:RETURN
410 DATA 07.0F,0F,07,0F,1C,1E,1F,3B,39,3C,3E,76,66,63,73
420 DATA 80,C0,C0,20,80,00,00,00,C0,C0,60,28,1C,3E,1E,8E
430 DATA 02,22,12,00,80,4F,14,24,44,7F,24,12,0A,05,02,01
440 DATA 40,44,48,00,01,F2,28,24,22,FE,24,48,50,A0,40,80
450 DATA 00,00,00,01,03,07,07,07,27,7F,FF,FE,1F,1E,1C
460 DATA 3F,7F,FF,FF,FF,FF,FE,FC,F8,F0,E0,00,00,00,00,00

```


古典的名作アクションゲーム

```

100 ' *** block2 ***
110 '
120 COLOR 15,1:SCREEN 1:WIDTH 29:KEY OFF
130 DEFSNG A-Z:R=RND(-TIME)
140 DEFFNR(A)=RND(1)*A
150 DEFFNP(X,Y)=VPEEK(6146+INT(X)+INT(Y)*32)
160 DEFFNB$(N)=STRING$(N,"0")
170 LF=5:XR=25:YR=10:LE=3
180 '
190 LOCATE 0,1:PRINT STRING$(25,"0")
200 FOR I=2 TO 20
210 PRINT "0";SPC(3);FNB$(2);SPC(4);FNB$(3)
220 NEXT I
230 PRINT STRING$(25,"0")
240 '
250 PLAY "L4N70":FOR W=0 TO 800:NEXT W
260 XB=13:YB=13:DX=1:DY=1
270 '
280 LOCATE XB,YB:PRINT "●"
290 DR=0:S=STICK(0):IF S=1 AND YR>1 THEN DR=-1
300 IF S=5 AND YR+LE<22 THEN DR=1
310 YR=YR+DR:LOCATE XR,YR-1:PRINT " ":FOR I=0 TO LE-1:LO
CATE XR,YR+I:PRINT "!":NEXT:LOCATE XR,YR+LE:PRINT " "
320 LOCATE 0,0:PRINT USING "SCORE #####";PO;:PRINT USING
" LEFT #":LF
330 LOCATE XB,YB:PRINT " ":X=XB+DX:Y=YB+DY
340 IF INT(Y)<2 OR INT(Y)>20 THEN DY=-DY
350 IF X<1 THEN DX=-DX
360 IF X<10 THEN IF SJ=0 THEN LE=LE-1:SJ=1
370 IF X>28 THEN LF=LF-1:IF LF<1 THEN PLAY "04L32BAGDC03
L8C":END ELSE 250
380 XB=XB+DX:YB=YB+DY:IF XB=XR AND INT(YB)=YR AND INT(Y
B)<YR+LE-1 THEN PLAY "L64N50":GOSUB 440
390 IF FNP(XB,YB)<>219 THEN 280
400 PO=PO+1:PLAY "L64N80":DX=-DX
410 IF FNP(XB+DX,YB+DY)=219 THEN DY=-DY
420 IF PO MOD 95=0 THEN PLAY "04L16BRBR3205AR807L4A":GOT
O 190 ELSE 280
430 '
440 DX=-DX:DY=FNR(2,1)-1:IF INT(YB)=YR THEN DY=FNR(1,1)-
2:RETURN
450 IF INT(YB)=YR+LE-1 THEN DY=FNR(1,1)+1
460 RETURN

```


8世紀の地球に侵入者あり!

36

平城京の闘い

ここは8世紀の平城京。わが銀河宇宙連合との泥沼化した戦いに決着をつけようとする敵ビボース帝国は、連合の中心である地球の過去を侵略してきたのだ!! 歴史を変えないためには、この世界で現代兵器は使えない。はたして歴史を守れるか!?




PLAY


隊長「この戦いは、かつてはやった『平安京エイリアン』とほとんど同じだ。が、注意しなければならないのは、穴掘りはスペースキーを押せばできるが、穴埋めはできないことだ。だから自分の作った穴にとじこめられないように。」

隊員A「しかし、それでは、エイリアンが穴に落ちてきたとき、埋められないのでは?」

隊長「大丈夫。やつらが穴に落ちると自動的に埋まってしまうのだ。とにかく制限時間内に全部のエイリアンをやつつけること。もちろんエイリアンとぶつかってもゲームオーバー。そして、残り時間がボナス点に関係するんで、なるべく早くやつつけることが至上命令である!!」




再プレイ……RUN (F-5)



PROGRAM

穴を掘るとき、画面に“5”，“4”，～“1”，“0”と表示されるが、これは、直接 VRAM の値を読み、空白ならば“5”をプリントする。キャラクターコードが49から53ならば、キャラクターコードを1つ減らして表示するというわけ。

エイリアンは“@”で表示しているが、これを変えたいときは、350行の最後の PRINT“@”を変えればよい。が、検非違使の“●”を変えるには、270行を変えるだけではダメ。340行の A = 133の部分も新しいキャラクターコードに変えること。



穴を掘っている間にエイリアンが近づいてきた

古典的名作アクションゲーム

```


100 '*** heijyo ***
110 '
120 COLOR 15,3:SCREEN 1:WIDTH 29:KEY OFF
130 DEFINT A-Y:R=RND(-TIME)
140 DEF FNP(X,Y)=VPEEK(6146+32*Y+X):DEF FNR(A)=RND(1)*A
150 DEF FNX(S)=VAL(MID$("112221000",S+1,1))-1
160 DEF FNY(S)=VAL(MID$("100122210",S+1,1))-1
170 '
180 LE=LE+1:CO=200:X=25:Y=21
190 FOR I=1 TO 8
200   X(I)=SGN(INT(LE*2/I)):Y(I)=3
210 NEXT I:CLS
220 FOR I=2 TO 22
230   LOCATE 0,I:PRINT STRING$(27,"□")
240 NEXT I
250 FOR I=1 TO 7:LOCATE 1,I*3:PRINT SPC(25):FOR J=3 TO 2
1:LOCATE I*4-3,J:PRINT " ":NEXT J,I
260 '
270 S=STICK(0):C=X:D=Y:GOSUB 400:IF J=32 THEN LOCATE X,Y
:PRINT " ":X=C:Y=D:LOCATE X,Y:PRINT "●"
280 IF STRIG(0)=0 THEN 300 ELSE C=X:D=Y:S=AN:GOSUB 400
290 IF J=32 THEN PLAY "L64N60":LOCATE C,D:PRINT "5" ELSE
IF J>48 AND J<54 THEN PLAY "L64N60":LOCATE C,D:PRINT CH
R$(J-1)
300 EJ=0
310 FOR I=1 TO 8
320   IF X(I)<1 THEN EJ=EJ+1:GOTO 360
330   PLAY "L64N20":IF Y(I)MOD3=0 AND X(I)MOD4=1 THEN S=
FNR(4)*2+1:A(I)=FNX(S):B(I)=FNY(S)
340   C=X(I)+A(I):D=Y(I)+B(I):A=FNP(C,D):IF A=219 THEN 3
60 ELSE IF A=133 THEN PLAY "L6404BAGFEDC":END
350   IF A=48 THEN PLAY "L6403CDEFG":LOCATE X(I),Y(I):PR
INT " ":PO=PO+100:LOCATE C,D:PRINT " ":X(I)=-5 ELSE LOCA
TE X(I),Y(I):PRINT " ":X(I)=C:Y(I)=D:LOCATE C,D:PRINT "@
"
360 NEXT I
370 CO=CO-1:GOSUB 440:IF EJ=8 THEN FOR I=1 TO CO:CO=CO-1
:PO=PO+10:PLAY "L64N40":GOSUB 440:FOR W=0 TO 50:NEXT W,I
:GOTO 180
380 IF CO>0 THEN 270 ELSE PLAY "L1A":END
390 '
400 C=C+FNX(S):D=D+FNY(S):J=FNP(C,D)
410 IF S>0 THEN AN=S
420 RETURN
430 '
440 LOCATE 5,0:PRINT USING "TIME ### " :CO::PRINT "SCORE
":PO:RETURN

```


37

迫りくるブロックインベーダーの恐怖 ブロック軍団の逆襲

宇宙にはまったく変な生物がいるものだ。いま地球を侵略しつつあるあのインベーダーは、まるでブロックの壁ではないか。などと感心している場合ではない。やつらは無感動に無機質に、じわじわと空から降りてくる。とにかく撃ちまくるんだ！



PLAY



ひとりでこのゲームをいうなら、インベーダータイプ。懐かしさで熱くなることウケアイだ。このブロック軍団は右から左に移動しながら、じわりと1段ずつおりてきてキミを脅かす。敵の移動はこれでなかなかスムーズだ。

敵の最上段が17段目にきたらゲームオーバー。それまでに砲台を左右に動かし、ミサイルを発射して敵をやっつけなければならない。敵はこちらのミサイルにあたったときだけ攻撃してくるが、これが実にどうして鋭い攻撃だ。うまくかわさないとやられてしまう。しかも、急いで敵を全滅させないと占領されて負けになってしまうぞ。

PROGRAM

1つ1つのブロックについて、生死の判断や位置をどう表示すべきか。配列に入れて、LOCATEとPRINT文を使って表示してもよいのだが、そうすると処理速度が非常に遅くなる。そこで考えた。1つの文字変数(ここでは1\$)に入れて、その表示位置を変えるのだ(420~430行)。こうすれば高速化を図ることができる。

<改良法> ブロックの数を増やしたいときは230行の1\$=~の中の60を大きくすればよい。ブロックが降りてくる速さを変えたければ、240行のINTERVAL=~の計算式を変えればよい。



撃たれたブロックが死に際に爆弾を落としたぞ

古典的名作アクションゲーム

```

100 ' *** bfblock ***
110 '
120 COLOR 15,1,7:SCREEN 1,,0:WIDTH 20:KEY OFF
130 CLEAR 1000:DEFINT A-Z:R=RND(-TIME):DIM S(2,13)
140 DEFFNP(X,Y)=VPEEK(6150+32*Y+X)
150 DEFFNL(X,Y)=(Y-IY)*20+(X-IX)+1
160 FOR J=1 TO 2
170   FOR I=0 TO 13
180     READ S(J,I)
190   NEXT I
200 NEXT J
210 LE=0
220 '
230 I$=STRING$(60,"0"):LE=LE+1:IY=5:CJ=0:PJ=0
240 ON INTERVAL=240/(6+LE) GOSUB 420
250 CLS:LOCATE IX,IY:PRINT I$:SOUND 7,248:PLAY "L6404CDE
FGAB":INTERVAL ON
260 '
270 INTERVAL STOP:LOCATE PX,21:PRINT " "
280 S=STICK(0):IF S=3 AND PX<19 THEN PX=PX+1
290 IF S=7 AND PX>0 THEN PX=PX-1
300 LOCATE PX,21:PRINT "+ "
310 IF PJ=0 THEN IF STRIG(0)=-1 THEN S=1:GOSUB 460:MX=PX
:MY=21:PJ=1 ELSE 360
320 LOCATE MX,MY:PRINT " ":MY=MY-1:IF MY<0 THEN PJ=0:GOT
O 360
330 IF FNP(MX,MY)<>>219 THEN LOCATE MX,MY:PRINT "I":GOTO
360
340 SOUND 7,248:PLAY "L64N80":MID$(I$,FNL(MX,MY),1)=" ":
LOCATE MX,MY:PRINT " ":PJ=0:PO=PO+1
350 IF I$=SPACE$(LEN(I$)) THEN 230 ELSE IF CJ=0 THEN CX=
MX:CY=MY:CJ=1
360 IF CJ=0 THEN 400
370 LOCATE CX,CY:PRINT " ":CY=CY+1
380 IF CX=PX AND CY=21 THEN S=2:GOSUB 460:END
390 IF CY>20 THEN CJ=0 ELSE LOCATE CX,CY:PRINT "●"
400 LOCATE 5,0:PRINT "Score ";PO:INTERVAL ON:GOTO 270
410 '
420 LOCATE IX,IY:PRINT " ":IX=IX+1:IF IX>19 THEN IX=0:IY
=IY+1
430 LOCATE IX,IY:PRINT I$:IF IY<=16 THEN RETURN
440 S=2:GOSUB 460:END
450 '
460 FOR L=0 TO 13:SOUND L,S(S,L):NEXT:RETURN
470 '
480 DATA 0.0.0.0.0.0.1,7,16,16,16,100,25,0
490 DATA 0.0.0.0.0.0,21,247,16,0,0,100,60,0

```

ビデオゲームの決定版、いまここに再現

38

テニスゲーム

家庭のTVの前にビデオゲームがあった数年前の光景が懐かしいが、いまTVの前にはMSXがある。ゲームのレベルはもちろん断然高くなった。でも、古き良きあのテニスゲーム—単純な球やラケットの動き—をMSXで再現したい！



PLAY




操作はかつてのビデオゲーム同様とても簡単。カーソルキーの上下でラケットを動かし、画面上をちょろちょろ動き回るボールを反射させるだけでいい。得点はミスするまで何回ラケットにあてられたかで決まる。単純な中にも、ラケットにボールがあたる位置ではね返る方向が微妙に変わるあたりは細かい。

そして、この画面を見てくれ。なんとリアルなボールやラケット。さすがMSX、これもスプライト機能のなせる技だ。

PROGRAM

プログラムは実に単純。ただし注意すべき点がいくつかある。たとえばボールとラケットの衝突はスプライトの衝突割り込みで処理しているのだが、割り込みの許可と禁止をどこでするのが問題だ。割り込みから返るときボールとラケットが離れていないと、再び割り込みがかかってしまうことがあるのだ。そこでここでは割り込みがかかるとすぐにSPRITE OFFを実行し(450行)、ボールが反対側の壁で反射したときSPRITE ONを実行している(380行)。

また、ラケットにぶつかったあとのボールの方向づけにも注目(460行)。Y方向の運動量が衝突前の方向と密接に関連している。だから場合によってはボールははね返ってY方向をグッと増すこともある。



リアルなラケットとボールに注目

古典的名作アクションゲーム

```

100 '*** tennis ***
110 '
120 COLOR 15,4:SCREEN 1,0:WIDTH 29:KEY OFF
130 DEFINT A-Z:R=RND(-TIME)
140 DEF FNR(X)=RND(1)*X-1
150 DEF FNY(S)=VAL(MID$("100122210",S+1,1))-1
160 FOR I=0 TO 2
170   S$=""
180   FOR J=1 TO 8
190     READ A$:S$=S$+CHR$(VAL("&H"+A$))
200   NEXT J
210   SPRITE$(I)=S$
220 NEXT I
230 PL=3
240 XB=96:YB=32:XR=240:YR=96:DX=8:DY=6:LV=3
250 CLS:PLAY "O4L64CDEF"
260 LOCATE 0,1:PRINT STRING$(28,"□")
270 FOR I=2 TO 21
280   PRINT "□"
290 NEXT I
300 PRINT STRING$(28,"□"):GOSUB 480
310 ON SPRITE GOSUB 440:SPRITE ON
320 '
330 FOR I=0 TO 1
340   PUT SPRITE I,(XR,YR+I*8).9,I
350 NEXT I
360 PUT SPRITE 2,(XB,YB),11,2
370 XB=XB+DX:YB=YB+DY
380 IF XB<16 THEN PLAY "C":DX=-DX:XB=XB+DX:SPRITE ON
390 IF XB>248 THEN PLAY "BAGFEDCB":PL=PL-1:GOSUB 480:FOR
   W=0 TO 3000:NEXT:IF PL=0 THEN END ELSE 240
400 IF YB<17 OR YB>168 THEN PLAY "G":DY=-DY:YB=YB+DY
410 S=STICK(0):DR=FNY(S)*4:YR=YR+DR:IF YR<16 OR YR>160 T
HEN YR=YR-DR
420 GOTO 330
430 '
440 IF XB<240 OR ABS(YB-YR-4)>12 THEN SPRITE ON:RETURN
450 SPRITE OFF:PLAY "BE":HT=HT+1:GOSUB 480
460 DX=-DX:DY=(DY+YB-YR-8)/2+FNR(3):RETURN
470 '
480 LOCATE 0,0:PRINT USING "Score #### Ball #":HT,PL:RE
TURN
490 '
500 DATA 38.6C,D6,AA,D6,AA,D6,AA
510 DATA D6,AA,D6,AA,D6,AA,54,38
520 DATA 00,3C,7E,7A,7A,66,3C,00

```


アクションゲーム38

MSXポケットバンク12

1984年8月20日 初版発行

定価480円

1985年6月15日 第1版第4刷発行

編著者 ぐるーぶ・アレフ

発行者 塚本慶一郎

発行所 株式会社 **アスキー**

〒107 東京都港区南青山5-11-5 住友南青山ビル5F

振替 東京4-161144

電話 03-486-7111 (代表)

©1984 ASCII Corporation. Printed in Japan.

本書は著作権法上の保護を受けています。本書の一部あるいは全部について（ソフトウェア及びプログラムを含む）、株式会社アスキーから文書による許諾を得ずに、いかなる方法においても無断で複写、複製することは禁じられています。

編集担当 八木淳一

表紙担当 郷 啓子

印刷 株式会社加藤文明社印刷所

ISBN4-87148-754-7 C0055 ¥480E

MSX POCKET BANK

音楽好き、
ゲーム好き、
アート好き。
それぞれうれしい
MSXポケットバンク。



各巻480円

- ① アニメC.G.に挑戦!
- ② マイコン・ジュークボックス
- ③ BASICゲーム教室
- ④ マイコン・サウンドバック
- ⑤ ゲームキャラクタ操縦法
- ⑥ トランプゲーム集
- ⑦ 面白パズルブック
- ⑧ プログラムD.J.
- ⑨ グラフィック秘伝
- ⑩ マイコン野球中継'84

11 とにかく速いマシン語ゲーム

とりえは速さだけではない。
面白さバツグンのマシン語
ゲームがキミを直撃!

MSX POCKET BANK 11
とにかく速い
マシン語ゲーム集



ポケット
バンク
編集部

12 アクションゲーム38

あつ、という間に打ちこめるオ
モシロゲームが38!!
ポケットバンクの出血大サー
ビス。

MSX POCKET BANK 12
アクションゲーム38



ぐるーぶ・
アレフ

13 知能ゲーム38

キミの頭は噴火寸前。
楽しく悩める知性派ゲームが
38本も直撃た。

MSX POCKET BANK 13
知能ゲーム38



ぐるーぶ・
アレフ

14 必殺ビデオ活用法

ビデオとMSXつなげたら、
10倍広がるビジュアル・ワー
ルド。今、君だけの映像を、
手に入れろ!!



ポケット
バンク
編集部

15 占っちゃうから!

世の果てからやって来た、不
思議占い!
おぼっちゃん、おじょうちゃ
んには、過激すぎるぜ。



ポケット
バンク
編集部

16 エラー撃退ミニ事典

迷える小羊に救いの道を・・・
プログラマのバグとり術を満
載。
君の悩みも、これで解決。



ポケット
バンク
編集部



MSXポケットバンク⑫
アクションゲーム38

CONTENTS

- ・〔マンガ版〕アクションゲームの作り方
- ・スプライト機能大図解
- ・グラフィック機能大図解
- ・〔マンガ版〕ゲームストーリー発想法講座
- ・ワンポイントレッスン
- ・ACTION GAME×38

MSXポケットバンク・シリーズ
好評既刊 各巻480円

- ①アニメC.G.に挑戦!
- ②マイコン・ジュークボックス
- ③BASICゲーム教室
- ④マイコン・サウンドパック
- ⑤ゲームキャラクタ操縦法
- ⑥トランプゲーム集
- ⑦面白パズルブック
- ⑧プログラムD.J.
- ⑨グラフィックス秘伝
- ⑩マイコン野球中継'84
- ⑪とにかく速いマシン語ゲーム集
- ⑫アクションゲーム38